

Mit Excel-Datum rechnen

Inhalt

Liste mit Datumswerten für mehrere Jahre erstellen	2
Datumszahl in korrektes Excel-Datum umwandeln	11
Anzahl von Jahrgängen aus einer Liste zählen	23
Mit BRTEILJAHRE() das Alter in Jahren berechnen	28
Monatsende eines bestimmten Monats berechnen	32
So ermitteln Sie das Quartal aus einem Datum	34
Datum in Kalenderwoche umwandeln mit Excel	37
Datum eines Wochentags aus Kalenderwoche berechnen	40
Anzahl der Feiertage zwischen zwei Datumswerten in Excel berechnen	42
So können Sie die Nettoarbeitstage in Excel berechnen	45
Mit ARBEITSTAG.INTL-Funktion ein Zieldatum berechnen	50
Liste mit Arbeitstagen automatisch erstellen	55
Arbeitstage für bestimmten Monat ausrechnen	58
Wie Sie Schaltjahre mit Excel berechnen	63
Umstellungsdatum für Sommer- und Winterzeit berechnen	66
Mit Datumsangaben vor 1900 in Microsoft Excel rechnen	68
Empfehlungen aus dem Management-Handbuch	74

Liste mit Datumswerten für mehrere Jahre erstellen

Wie Sie mit Power Query eine Liste mit allen Tagen für mehrere Jahre erstellen – beginnend mit dem 1.1. des Jahres des frühesten Datums und endend mit dem 31.12. des Jahres mit dem letzten Datum der ursprünglichen Liste.

Zuletzt geändert am 11.03.2026



Ausgangssituation: Tabelle mit Datumsangaben und Werten

Sie haben eine Tabelle mit Werten für unterschiedliche Tage. Das Datum und die Werte stehen jeweils in einer Spalte untereinander. Allerdings ist die Liste nicht durchgängig, nicht jeder Tag ist aufgeführt.

	A	B
1	Datum	Wert
2	03.02.2023	144
3	05.02.2023	110
4	26.03.2024	139
5	15.04.2024	107
6	23.06.2024	146
7	12.07.2024	108
8	18.07.2024	120
9	08.08.2024	108
10	12.08.2024	105
11	22.08.2024	104
12	31.08.2024	109
13	12.09.2024	124
14	17.10.2024	110
15	29.10.2024	118
16	10.11.2024	148
17	13.12.2024	120
18	02.01.2025	155
19	13.02.2025	124
20	18.06.2025	178
21	19.09.2025	165
22	30.12.2025	101

Liste mit Angaben zum Datum und entsprechenden Werten


Für eine Auswertung der Daten benötigen Sie eine Liste, in der jeder Tag einzeln in der Spalte **Datum** aufgeführt ist. Dabei sollen alle Jahre, für die ein Datum in Ihrer Liste steht, vom 01.01. bis zum 31.12. aufgeführt sein.

Ist der erste Eintrag beispielsweise der 03.02.2023 und der letzte Wert der 30.12.2025, dann soll die Datumsliste vom 01.01.2023 bis zum 31.12.2025 reichen.

Eine solche Liste erstellen Sie mit Power Query.

Liste mit Datum und Wert in Power Query importieren

Importieren Sie zunächst Ihre bestehende Liste in Power Query. Markieren Sie eine Zelle in der Liste und aktivieren Sie im Menüband die Registerkarte **Daten** > Befehlsgruppe **Daten abrufen und transformieren** > Befehl **Aus Tabelle/Bereich**.



Daten aus Excel in Power Query importieren


Da es sich bei der Liste vermutlich um einen normalen Bereich handelt, wird die Liste zunächst in eine intelligente Tabelle umgewandelt.

Es öffnet sich das Dialogfeld **Tabelle erstellen**. Der Bereich der Liste wird automatisch erkannt und markiert.

Da die Liste über eine Beschriftung verfügt, muss das Kontrollkästchen **Tabelle hat Überschriften** aktiviert sein.

Bestätigen Sie die Einstellungen durch einen Klick auf **OK**.

	A	B	C	D	E
1	Datum	Wert			
2	03.02.2023	144			
3	05.02.2023	110			
4	26.03.2024	139			
5	15.04.2024	107			
6	23.06.2024	146			
7	12.07.2024	108			
8	18.07.2024	120			
9	08.08.2024	108			
10	12.08.2024	105			
11	22.08.2024	104			
12	31.08.2024	109			
13	12.09.2024	124			
14	17.10.2024	110			
15	29.10.2024	118			
16	10.11.2024	148			
17	13.12.2024	120			
18	02.01.2025	155			
19	13.02.2025	124			
20	18.06.2025	178			
21	19.09.2025	165			
22	30.12.2025	101			



Einstellungen zu den zu importierenden Daten

Die Liste wird daraufhin in den Power-Query-Editor geladen. Dort ist im Folgenden der Name der importierten Tabelle geändert zu **Datumstabelle**.

Doppelklicken Sie dazu den ursprünglichen Namen in der linken Spalte und ändern Sie den Namen entsprechend.

Anschließend ändern Sie noch den **Datentyp** der ersten Spalte mit dem Datum in das Format **Datum**. Dazu klicken Sie das kleine Symbol im Spaltentitel links und wählen den entsprechenden Datentyp

Abfragen [1]

Datumstabelle

= Table.TransformColumnTypes(Quelle,{{"Datum", type date}})

	Datum	Wert
1	03.02.2023	144
2	05.02.2023	110
3	26.03.2024	139
4	15.04.2024	107
5	23.06.2024	146
6	12.07.2024	108
7	18.07.2024	120
8	08.08.2024	108
9	12.08.2024	105
10	22.08.2024	104
11	31.08.2024	109
12	12.09.2024	124
13	17.10.2024	110
14	29.10.2024	118
15	10.11.2024	148
16	13.12.2024	120
17	02.01.2025	155
18	13.02.2025	124
19	18.06.2025	178
20	19.09.2025	165
21	30.12.2025	101

Name der Tabelle (Abfrage) in Power Query und Datentyp ändern

Neue Abfrage erstellen

Um nun eine Tabelle mit allen Datumswerten für die jeweiligen Jahre zu erzeugen, definieren Sie zunächst eine neue Abfrage. Führen Sie dazu im Menüband die Befehlsfolge Registerkarte **Start** > **Neue Quelle** > **Andere Quellen** > **Leere Abfrage** aus.

Alternative: Sie klicken im linken, leeren Bereich mit der rechten Maustaste und wählen aus dem Kontextmenü **Neue Abfrage** > **Andere Quellen** > **Leere Abfrage**.

The screenshot shows the Power Query Editor interface. The 'Datei' tab is active, displaying a ribbon with options like 'Schließen & laden', 'Vorschau aktualisieren', 'Eigenschaften', 'Erweiterter Editor', 'Verwalten', 'Spalten auswählen', 'Spalten entfernen', 'Spalten verwalten', 'Zeilen beibehalten', 'Zeilen entfernen', 'Zeilen verringern', 'Sortieren', 'Spalte teilen', 'Gruppieren nach', 'Datentyp: Datum', 'Erste Zeile als Übers', and 'Werte ersetzen'. The main area shows a table with two columns: 'Datum' and 'Wert'. The 'Datum' column contains dates from 03.02.2023 to 30.12.2025. The 'Wert' column contains numerical values ranging from 101 to 144. A context menu is open over the 'Datei' tab, with 'Neue Abfrage' and 'Leere Abfrage' highlighted. The formula bar at the top displays the M-code: `= Table.TransformColumnTypes(Quelle,{{"Datum", type date}})`.

	Datum	Wert
1	03.02.2023	144
2	05.02.2023	110
3	26.03.2024	139
	15.04.2024	107
		146
		108
		120
		108
9		
10		
11		
12	12.09.2024	
13	17.10.2024	
14	29.10.2024	118
15	10.11.2024	148
16	13.12.2024	120
17	02.01.2025	155
18	13.02.2025	124
19	18.06.2025	178
20	19.09.2025	165
21	30.12.2025	101

Power-Query-Funktion für eine leere, neue Abfrage

M-Funktion für die Liste aller Tage (Datumswerte)

Um alle Tage für die betroffenen Jahre in einer Liste (Spalte) zu erstellen, müssen Sie eine Datumsreihe generieren. Nutzen Sie hierzu folgenden Code in der Befehlszeile (M-Funktion):


```
= List.Dates(Date.StartOfYear (List.Min (Table.Column(Datumstabelle, "Datum"))), Number.From(Date.EndOfYear (List.Max (Table.Column(Datumstabelle, "Datum")))) - Number.From(Date.StartOfYear (List.Min(Table.Column(Datumstabelle, "Datum")))) + 1 , #duration(1,0,0,0))
```

- Der Parameter **Datumstabelle** ist die Tabelle (Quelle), in der die Datumswerte stehen, die Sie betrachten.
- Der Parameter **Datum** ist der Name der Spalte, in der die Datumswerte stehen.

Der Code sucht in der Spalte Datum nach dem kleinsten Wert, dem ersten eingetragenen Datum (**List.Min**) und nimmt dann den Wert für den Beginn des Jahres dieses Datums (**Date.StartOfYear**), also der 01.01. des gefundenen Jahres.

Dann sucht der Code den größten Datumswert (**List.Max**) und ermittelt das Ende des entsprechenden Jahres (**Date.EndOfYear**), also der 31.12. des gefundenen Jahres. Von diesem Wert wird der Wert des 01.01. des ersten Jahres über die Funktion **Number.From** abgezogen, sodass die Zahl der Tage zwischen dem 01.01. des ersten Jahres und dem 31.12. des letzten Jahres berechnet wird.

Mit **List.Dates()** werden dann alle Datumswerte für jeden Tag dazwischen erzeugt, indem mit **#duration** so viele Tage aufgelistet werden, wie als Zahl berechnet wurden.




Alle Tage eines Jahres als Liste in Power Query

Liste in Tabelle verwandeln

Wandeln Sie die Liste jetzt in eine Tabelle um, indem Sie im Menüband die Befehlsfolge **Listentools** > **Transformieren** > **Zu Tabelle** ausführen.

Alternativ: Sie klicken mit der rechten Maustaste auf den Spaltenkopf **Liste** und wählen aus dem Kontextmenü **Zu Tabelle**.

Es öffnet sich das Dialogfeld **Zu Tabelle**. Lassen Sie die Einstellungen unverändert und bestätigen Sie das Dialogfeld durch einen Klick auf **OK**.



Liste in eine Tabelle umwandeln

Liste mit Datumswerten nach Excel übergeben

Dann übergeben Sie Ihre Liste wieder an Ihr Tabellenblatt in Excel, indem Sie im Menüband die Befehlsfolge **Start > Schließen & laden** wählen.

Sie haben nun eine (lange) Liste mit dem Datum für jeden Tag der Jahre, die in Ihrer ursprünglichen Liste aufgeführt sind.

	A
1	Datum
2	01.01.2023
3	02.01.2023
4	03.01.2023
5	04.01.2023
6	05.01.2023
7	06.01.2023
8	07.01.2023
9	08.01.2023
10	09.01.2023
11	10.01.2023
12	11.01.2023
13	12.01.2023
14	.
15	.
16	.
17	.
1091	25.12.2025
1092	26.12.2025
1093	27.12.2025
1094	28.12.2025
1095	29.12.2025
1096	30.12.2025
1097	31.12.2025

Ergebnis: Liste mit allen Datumswerten (Ausschnitt Anfang und Ende der Liste)



Tipp: Datum und weitere Einträge der Ursprungstabelle ergänzen

In der Ausgangstabelle in Excel ist neben dem Datum eine weitere Spalte mit Werten angegeben. Wenn Sie die Einträge aus dieser Spalte ebenfalls ergänzen wollen, sodass jedes Datum eines Jahres auch einen Wert hat, gehen Sie so vor, **wie in dieser Excel-Anleitung beschrieben**.

Datumszahl in korrektes Excel-Datum umwandeln

Ist ein Datum ohne Punkt oder anderes Zeichen nur als Zahl angegeben, kann Excel daraus nicht unmittelbar das korrekte Datum ermitteln. Sie können dennoch eine Datumszahl wie 2802 in ein richtiges Datum umwandeln. So gehen Sie vor.

Zuletzt geändert am 11.03.2026



Beim Import von Daten in Power Query kann es vorkommen, dass Datumswerte in einem verkürzten Format vorliegen.

Ein Beispiel ist die Darstellung **2802** für den 28. Februar 2025 oder **103** für den 1. März 2025. In diesem Excel-Tipp erfahren Sie, wie Sie diese Werte in das richtige Datumsformat umwandeln.

Beispiel: Spalte mit Zahlen in Power Query


In der folgenden Abbildung sehen Sie eine Abfrage in Power Query, die als Beispiel dienen soll. Datumswerte werden hier verkürzt als einfache Zahl dargestellt; dabei ist eigentlich ein Datum gemeint.

	1 ² Datum
1	2802
2	103
3	1504
4	3112
5	101

Zahlen, die eigentlich ein Datum meinen


Bei Import wurde automatisch der Datentyp **Ganze Zahl** ausgewählt. Dies führt dazu, dass führende Nullen nicht dargestellt werden. Wandeln Sie daher den Datentyp zunächst wieder in **Text** um.

Klicken Sie am linken Rand der Spalte auf das Symbol für den Datentyp (hier 123) und wählen Sie in der Liste den Datentyp **Text** aus.



Werte in Datentyp Text umwandeln

Es öffnet sich das Dialogfeld **Spaltentyp ändern**. Klicken Sie hier auf die Schaltfläche **Aktuelle ersetzen**, um den aktuellen Datentyp in **Text** zu ersetzen.



Spaltentyp nochmals ändern – und aktuellen ersetzen

Die Werte sind in den Datentyp Text umgewandelt. Dies führt dazu, dass jetzt wieder führende Nullen angezeigt werden.

Hinweis: Dabei ist vorausgesetzt, dass in der ursprünglichen Datenquelle das Datum so eingegeben wurde, dass Tage mit einer Ziffer (1. bis 9. eines Monats) immer mit der führenden 0 eingegeben sind. Sie erreichen dies dadurch, dass Sie das benutzerdefinierte Zahlenformat 0000 wählen. Wie Sie dies einstellen, erfahren Sie in diesem [Excel-Tipp](#).


	A ^B C Datum
1	2802
2	0103
3	1504
4	3112
5	0101

Zahlen mit Datentyp Text formatiert

Aus diesen verkürzten Datumswerten können Sie die „richtigen“ Datumswerte ableiten, indem Sie die Tages- und Monatswerte in separaten Spalten ableiten.

Tag aus Datumszahl extrahieren

Als Erstes berechnen Sie die Tageszahl aus diesem Datumswert. Aktivieren Sie im Menüband die Befehlsfolge **Spalte hinzufügen > Benutzerdefinierte Spalte**.



Benutzerdefinierte Spalte in Power Query einfügen

Es öffnet sich das Dialogfeld **Benutzerdefinierte Spalte**. Vergeben Sie hier unter **Neuer Spaltenname** einen aussagekräftigen Namen für die Spalte; zum Beispiel **Tag**, da Sie in dieser Spalte die Tageszahl ausweisen wollen.

Bei **Benutzerdefinierte Spaltenformel** erfassen Sie die folgende Formel:

=Text.Start([Datum], 2)

Mit dieser Formel extrahieren Sie die ersten zwei Zeichen aus der Spalte Datum. Aus dem Wert **2802** wird somit der Wert **28** usw.

Benutzerdefinierte Spalte

Fügen Sie eine Spalte hinzu, die aus den anderen Spalten berechnet wird.

Neuer Spaltenname
Tag

Benutzerdefinierte Spaltenformel ⓘ
= Text.Start([Datum],2)

Verfügbare Spalten
Datum

<< Einfügen

Weitere Informationen zu Power Query-Formeln

✓ Es wurden keine Syntaxfehler erkannt.

OK Abbrechen

Power-Query-Funktion zur Auswahl der ersten beiden Zeichen aus einem Text

Es wird eine neue Spalte mit dem Namen **Tag** eingefügt, in welcher die ersten beiden Zeichen aus der Spalte Datum dargestellt werden. Sie erhalten somit die **Tageszahl** aus dem Datumswert.

Abfragen [1]

Tabelle1

= Table.AddColumn("#Geänderter Typ",

	AB _C Datum	ABC 123 Tag
1	2802	28
2	0103	01
3	1504	15
4	3112	31
5	0101	01

Ergebnis: Neue Spalte mit Tageszahl des Datums

Hinweis: Wenn in der Spalte Datum die führende Null fehlt und zum Beispiel 103 für den 01.03.2025 erfasst wurde, dann geben Sie als benutzerdefinierte Spaltenformel folgenden Befehl ein:

= if Text.Length([Datum]) = 3 then Text.Start([Datum], 1) else Text.Start([Datum], 2))

Monat aus Datumszahl extrahieren

Wiederholen Sie nun diesen Vorgang für den **Monatswert**. Erstellen Sie eine weitere benutzerdefinierte Spalte und verwenden Sie nun die folgende Formel:

=Text.End([Datum], 2)

Mit dieser Formel leiten Sie aus der Spalte Datum die letzten beiden Zeichen ab, die der **Monatszahl** aus dem Datum entsprechen.

Benutzerdefinierte Spalte

Fügen Sie eine Spalte hinzu, die aus den anderen Spalten berechnet wird.

Neuer Spaltenname

Monat

Benutzerdefinierte Spaltenformel ⓘ

= Text.End([Datum], 2)

Verfügbare Spalten

Datum

Tag

<< Einfügen

Weitere Informationen zu Power Query-Formeln

✓ Es wurden keine Syntaxfehler erkannt.

OK Abbrechen

Power-Query-Funktion zur Auswahl der letzten beiden Zeichen aus einem Text

Sie haben nun eine weitere Spalte mit dem Namen **Monat** eingefügt, in der die Monatszahl aus der Spalte Datum steht.

Abfragen [1]

Tabelle1

Ergebnis: Neue Spalte mit der Monatszahl des Datums

Datum aus den neuen Spalten zusammenstellen

Mit diesen Nebenrechnungen (Spalten) können Sie nun ein vollständiges Datum ableiten. Erstellen Sie nun wieder eine neue benutzerdefinierte Spalte und erfassen Sie in dieser die folgende Formel:

=[Tag] & "." & [Monat] & "." & "2025"

Die abgeleiteten Werte aus der Spalte **Tag** und **Monat** und der Jahreszahl **2025** werden durch die **&-Zeichen** verkettet. Zwischen den Werten wird ein Punkt als Trennzeichen eingefügt. Das &-Zeichen ist eine praktische und einfache Alternative zur M-Funktion **Text.Combine**.

Hinweis: Wenn das Datum für ein anderes Jahr gilt, ändern Sie den letzten Parameter "2025" entsprechend. Sie können den Jahreswert auch aus einer anderen vorhandenen Spalte beziehen, wenn dort das Jahr eingetragen ist.

Benutzerdefinierte Spalte

Fügen Sie eine Spalte hinzu, die aus den anderen Spalten berechnet wird.

Neuer Spaltenname
Datum neu

Benutzerdefinierte Spaltenformel ⓘ
= [Tag] & " " & [Monat] & " " & "2025"

Verfügbare Spalten
Datum
Tag
Monat

<< Einfügen

Weitere Informationen zu Power Query-Formeln

✓ Es wurden keine Syntaxfehler erkannt.

OK Abbrechen

Power-Query-Funktion zur Verknüpfung von Spalten in einer neuen Spalte

Es wird eine benutzerdefinierte Spalte eingefügt, in der die einzelnen Datumsbestandteile hintereinander gereiht werden.

Die Werte sehen jetzt wie ein Datum aus, werden aber von Power Query noch als Text behandelt. Sie können dies daran erkennen, dass die Werte linksbündig in der Zelle dargestellt sind.

Abfragen [1]
Tabelle1


= Table.AddColumn("#Hinzugefügte benutzerdefinierte Spalte1", "Datum neu", each [Tag]

	ABC Datum	ABC Tag	ABC Monat	ABC Datum neu
1	2802	28	02	28.02.2025
2	0103	01	03	01.03.2025
3	1504	15	04	15.04.2025
4	3112	31	12	31.12.2025
5	0101	01	01	01.01.2025

Ergebnis: Neue Spalte mit den Angaben für ein korrektes Datum


Weisen Sie der Spalte nun als Nächstes den Datentyp **Datum** zu.

Klicken Sie hierzu links in der Spaltenbeschriftung auf das Symbol für den Datentyp (hier ABC123) und wählen Sie in der Liste den Eintrag **Datum** aus.



Datenformat Datum in Power Query auswählen

Nun werden die Werte in der Spalte als Datum erkannt. Sie können dies daran erkennen, dass die Werte nun **rechtsbündig** in der Zelle ausgerichtet sind, und als Datentyp ist in der Spaltenüberschrift links **Datum** eingestellt.



Ergebnis: Spalte als Datum formatiert

Alle Einzelschritte und Funktionen in Power Query kombinieren

Diese Schritte können Sie auch in einem Schritt durchführen, indem Sie auf Hilfsspalten verzichten und alle Schritte in einer benutzerdefinierten Spalte durchführen. Die Formel für die Spalte lautet dann:

=Text.Start([Datum], 2) &". " &Text.End([Datum], 2) &". " &"2025"

Benutzerdefinierte Spalte

Fügen Sie eine Spalte hinzu, die aus den anderen Spalten berechnet wird.

Neuer Spaltenname
Datum neu

Benutzerdefinierte Spaltenformel ⓘ
= Text.Start([Datum],2)&"."&Text.End([Datum],2)&"."&"2025"

Verfügbare Spalten
Datum

<< Einfügen

Weitere Informationen zu Power Query-Formeln

✓ Es wurden keine Syntaxfehler erkannt.

OK Abbrechen

Alle Power-Query-Funktionen zur Zusammenstellung des Datums in einer Funktion

Mit dieser Formel können Sie alles in einer einzigen Spalte abwickeln. Sie sparen sich somit die beiden Hilfsspalten für Tag und Monat. Allerdings sind die Werte auch noch vom Typ **Text**.

Abfragen [2] <

Tabelle1


Tabelle1 (2)

fx = Table.AddColumn(Quelle, "Datum neu",

	ABC 123 Datum	ABC 123 Datum neu
1	2802	28.02.2025
2	0103	01.03.2025
3	1504	15.04.2025
4	3112	31.12.2025
5	0101	01.01.2025


Ergebnis: Zahl zerlegen und als Datum zusammenfügen – in einem Schritt

Wandeln Sie die Textwerte jetzt noch in Datumswerte um, indem Sie den Datentyp auf **Datum** umstellen.



Power-Query-Spalte als Datum formatieren

Jetzt werden die Werte richtig als Datum erkannt und dargestellt.



Ergebnis: Aus einer Zahl ein Datum erstellt

Mit diesen einfachen Schritten können Sie verkürzte Datumswerte in Power Query problemlos in echte Datumswerte umwandeln. Dies erleichtert die Weiterverarbeitung in Excel und anderen Anwendungen.

Dazu übergeben Sie die Power-Query-Tabelle wieder einem Excel-Arbeitsblatt mit der Befehlsfolge **Start > Schließen & laden**.

Anzahl von Jahrgängen aus einer Liste zählen

Wie Sie ohne Umwege aus einer Liste mit Datumswerten die Zahl der Einträge für ein bestimmtes Jahr ermitteln. So sind beispielsweise die Geburtsdaten von Mitarbeitenden schnell ausgezählt für eine Altersübersicht.

Zuletzt geändert am 11.03.2026



Die **Aufgabe** besteht darin, auf der Basis einer Geburtstagsliste die Geburtsjahrgänge aus bestimmten Jahren zu zählen. In der folgenden Abbildung finden Sie ein entsprechendes Beispiel hierzu.

In den Spalten A und B sehen Sie eine Personalliste mit den entsprechenden Geburtsdaten. Im Bereich D2:E6 soll für die angegebenen Geburtsjahre ausgerechnet werden, wie viele Personen im entsprechenden Jahr geboren sind.

	A	B	C	D	E
1	Name	Geburtstag		Jahr	Anzahl
2	Thomas	09.28.1993		1991	
3	Erich	12.09.1992		1992	
4	Sabine	03.30.1993		1993	
5	Claus	03.01.1994		1994	
6	Timon	12.06.1991		1995	
7	Sonja	10.10.1992			
8	Julia	10.21.1995			
9	Daniel	04.04.1991			
10	Christian	05.25.1994			
11	Mike	10.13.1992			
12	Verena	09.03.1995			
13	Philipp	03.21.1991			

Liste mit Datumswerten

Bedingte Summe mit Matrixformel

Diese Aufgabe können Sie mit der folgenden Matrixformel lösen:

{=SUMME(WENN(JAHR(\$B\$2:\$B\$13) = D2; 1; 0))}

Erfassen Sie die Formel ohne die geschweiften Klammern am Anfang und Ende. Bestätigen Sie die Formel aber nicht mit Enter, sondern mit der Tastenkombination **Strg + Umschalttaste + Enter**, da es sich um eine Matrixformel handelt.

Haben Sie ein Microsoft 365-Abo, dann können Sie die Formel auch nur mit **Enter** erfassen.

E2 : {=SUMME(WENN(JAHR(\$B\$2:\$B\$13)=D2;1;0))}					
	A	B	C	D	E
1	Name	Geburtstag		Jahr	Anzahl
2	Thomas	09.28.1993		1991	3
3	Erich	12.09.1992		1992	3
4	Sabine	03.30.1993		1993	2
5	Claus	03.01.1994		1994	2
6	Timon	12.06.1991		1995	2
7	Sonja	10.10.1992			
8	Julia	10.21.1995			
9	Daniel	04.04.1991			
10	Christian	05.25.1994			
11	Mike	10.13.1992			
12	Verena	09.03.1995			
13	Philipp	03.21.1991			

Matrixformel zur bedingten Summierung von Werten

Was bewirkt diese Formel?

Zunächst wird aus den Datumswerten im Bereich B2:B13 die jeweilige Jahreszahl mit **JAHR()** extrahiert und anschließend mit dem Zellwert in der Zelle D2 verglichen.

{=SUMME(WENN(JAHR(\$B\$2:\$B\$13) = D2; 1; 0))}

Entspricht die Jahreszahl dem Wert in der Zelle D2, dann wird der Wert 1, sonst der Wert 0 über die Funktion WENN() an die Funktion SUMME() übergeben.

{=SUMME(WENN(JAHR(\$B\$2:\$B\$13) = D2; 1; 0))}

Dann werden mit SUMME() alle übergebenen Werte (0 oder 1) aufsummiert.

{=SUMME(WENN(JAHR(\$B\$2:\$B\$13) = D2; 1; 0))}

So bekommen Sie alle entsprechenden Jahrgänge in der jeweiligen Zeile.

Alternative über SUMMENPRODUKT()

Auch mit der Funktion **SUMMENPRODUKT()** können Sie diese Aufgabe schnell und einfach lösen. Erfassen Sie hierzu die folgende Formel in der Zelle E2:

=SUMMENPRODUKT((JAHR(\$B\$2:\$B\$13) = D2) *1)

	A	B	C	D	E
1	Name	Geburtstag		Jahr	Anzahl
2	Thomas	09.28.1993		1991	3
3	Erich	12.09.1992		1992	3
4	Sabine	03.30.1993		1993	2
5	Claus	03.01.1994		1994	2
6	Timon	12.06.1991		1995	2
7	Sonja	10.10.1992			
8	Julia	10.21.1995			
9	Daniel	04.04.1991			
10	Christian	05.25.1994			
11	Mike	10.13.1992			
12	Verena	09.03.1995			
13	Philipp	03.21.1991			

Bedingte Summierung mit SUMMENPRODUKT()

Hier wird zunächst über die Funktion **JAHR()** die jeweilige Jahreszahl aus den Datumswerten im Bereich B2:B13 ermittelt und mit dem Wert in der Zelle D2 verglichen. Stimmen der Wert überein, dann wird eine 1, sonst eine 0 zurückgegeben.

=SUMMENPRODUKT((JAHR(\$B\$2:\$B\$13) = D2) *1)

Anschließend werden die einzelnen Werte mit 1 multipliziert. Dadurch werden nur die Einträge berücksichtigt, für welche die Prüfung vorher den Wert „wahr“ oder 1 ermittelt hat.

=SUMMENPRODUKT((JAHR(\$B\$2:\$B\$13) = D2) *1)

Diese Werte der Multiplikation werden schließlich summiert. Auf diese Weise erhalten Sie auch die einzelnen Summen der Jahrgänge.

=SUMMENPRODUKT((JAHR(\$B\$2:\$B\$13) = D2) *1)

Mit BRTEILJAHRE() das Alter in Jahren berechnen

Für die Personalstatistik braucht es das Alter der Beschäftigten. Für die Kapazitätsplanung das Alter von Maschinen und Anlagen. Mit der Excel-Funktion BRTEILJAHRE() können Sie das Alter in Jahren aus zwei Datumswerten berechnen.

Zuletzt geändert am 11.03.2026



In der betrieblichen Praxis werden Sie regelmäßig mit der Aufgabe konfrontiert, dass Sie das **Alter von Personen, Maschinen oder Anlagen** berechnen müssen.

Grundsätzlich können Sie hierfür die Funktion DATEDIF() verwenden. DATEDIF() ist allerdings keine offizielle Funktion von Excel. Sie finden diese Funktion auch nicht im Funktionsassistenten.

Eine Alternative ist eine relativ unbekannte Funktion für diese Aufgabe. Es handelt sich hierbei um die Funktion **BRTEILJAHRE()**.

Parameter der Excel-Funktion BRTEILJAHRE()

Die Funktion BRTEILJAHRE() wurde mit Excel 2007 neu eingeführt und berechnet den Teil in Jahren, der durch die Anzahl von ganzen Tagen zwischen zwei Datumsangaben dargestellt wird. BRTEILJAHRE() besitzt die folgende Syntax:

=BRTEILJAHRE(Ausgangsdatum; Enddatum; [Basis])

Dabei gilt:

- **Ausgangsdatum:** Datum, welches das Ausgangsdatum für die Berechnung angibt. Im Gegensatz zur Funktion DATEDIF() kann dieses Datum auch größer sein, als das Enddatum. Die Funktion BRTEILJAHRE() liefert auch für diesen Fall das richtige Ergebnis.
- **Enddatum:** Datum, welches das Enddatum für die Berechnung angibt.
- **Basis** (optional): Hier definieren Sie, wie die Tage gezählt werden.

Für den Wert **Basis** gibt es unterschiedliche Systematiken – abhängig davon, mit wie vielen Tagen pro Jahr oder pro Monat gerechnet wird. Als Wert für Basis können Sie entsprechend angeben:

- oder nicht angegeben – USA (NASD) 30/360
- **1** – actual/actual
- **2** – actual/360
- **3** – actual/365
- **4** – Europa 30/360

Hinweis: Basis für die Berechnung des Alters

Je nach Anwendungszweck wird die Zahl der Tage pro Monat pauschal mit 30 und die Zahl der Tage pro Jahr pauschal mit 360 angenommen. Für die Formel BRTEILJAHRE() gilt für den Wert **Basis**:

- vor dem Schrägstrich (/) – Zahl der Tage pro Monat, mit der gerechnet wird; actual ist die tatsächliche Zahl an Tagen zwischen Ausgangsdatum und Enddatum.
- nach dem Schrägstrich (/) – Zahl der Tage pro Jahr, mit der gerechnet wird; actual ist die tatsächliche Zahl an Tagen zwischen Ausgangsdatum und Enddatum.

Die exakte Altersangabe, bei der die genaue Zahl an Tagen pro Monat und die Schaltjahre berücksichtigt werden, wird also bei der Systematik **1 = actual/actual** berechnet.

Beispiel für BRTEILJAHRE()

Die Anwendung der Funktion **BRTEILJAHRE()** soll an einem praktischen **Beispiel** illustriert werden. Das Alter von Vermögensgegenständen soll zum Bilanzstichtag 31.12.2021 berechnet werden.

Erfassen Sie daher als Funktionsargumente das Anschaffungsdatum und das Datum des Bilanzstichtages. Für die Basis erfassen Sie den Wert 4, damit das europäische Zählmodell angewendet wird, wie es bei der Bilanzanalyse üblich ist.

C2						
=BRTEILJAHRE(A2;B2;4)						
	A	B	C	D	E	
1	Anschaffung	Stichtag	Alter			
2	01.04.1974	31.12.2021	47,75			
3	10.11.2005	31.12.2021	16,14			
4	07.08.1999	31.12.2021	22,40			
5	31.12.2020	31.12.2021	1,00			
6						

Beispiel für die Excel-Funktion BRTEILJAHRE()

Alter auf volle Jahre runden

BRTEILJAHRE() stellt Ihnen grundsätzlich die noch nicht vollendeten Jahre als Nachkommastellen dar. Wollen Sie das Alter in ganzen Jahren darstellen, dann können Sie die Funktion BRTEILJAHRE() in die Funktion **ABRUNDEN()** verschachteln.

Im Beispiel müssen Sie daher die Formel in der Zelle C2 wie folgt anpassen:

=ABRUNDEN(BRTEILJAHRE(A2;B2; 4); 0)

Im ersten Argument der Funktion ABRUNDEN() erfassen Sie die Funktion BRTEILJAHRE() und im zweiten Argument vergeben Sie hier den Wert 0. Damit stellen Sie ein, dass das Ergebnis der Funktion BRTEILJAHRE() auf die nächste ganze Zahl abgerundet wird.

C2							
=ABRUNDEN(BRTEILJAHRE(A2;B2;4);0)							
	A	B	C	D	E	F	
1	Anschaffung	Stichtag	Alter				
2	01.04.1974	31.12.2021	47,00				
3	10.11.2005	31.12.2021	16,00				
4	07.08.1999	31.12.2021	22,00				
5	31.12.2020	31.12.2021	1,00				
6							

Berechnung des Alters auf volle Jahre abrunden

Monatsende eines bestimmten Monats berechnen

Der letzte Tag in drei Monaten – welches Datum hat dieser letzte Tag? Und auf welchen Wochentag fällt dann das Monatsende? Diese Frage beantworten Sie mithilfe der Excel-Funktion **MONATSENDE()**.

Zuletzt geändert am 11.03.2026



Mit der Funktion **MONATSENDE()** können Sie in Excel den letzten Tag eines bestimmten Monats berechnen.

Hierfür müssen Sie lediglich einen Datumswert als Ausgangsdatum erfassen, um ein entsprechendes Monatsende in Excel bestimmen zu können.

Dies funktioniert nicht nur für den Monat des Ausgangsdatums, sondern auch für Monate, die vor oder nach dem Ausgangsdatum liegen.

Die Funktion hat folgende Syntax:

=MONATSENDE(Ausgangsdatum; Monate)

- **Ausgangsdatum:** Datumswert, der das Startdatum darstellt.
- **Monate:** Gibt an, wie viele Monate vor oder nach dem Ausgangsdatum liegen sollen. Ein positiver Wert für Monate ergibt ein in der Zukunft liegendes Datum, ein negativer Wert ein in der Vergangenheit liegendes Datum.

C2						
	A	B	C	D	E	F
1	Ausgangsdatum	Monate	Ergebnis	Ergebnis im Format TTTT	Formel	Beschreibung
2	01.01.2024	-1	31.12.2023	Sonntag	=MONATSENDE(A2;B2)	Monatsende des Vormonats
3	27.03.2024	0	31.03.2024	Sonntag	=MONATSENDE(A3;B3)	Monatsende des aktuellen Monats
4	01.01.2024	1	29.02.2024	Donnerstag	=MONATSENDE(A4;B4)	Monatsende den nächsten Monats
5	15.01.2024	2	31.03.2024	Sonntag	=MONATSENDE(A5;B5)	Monatsende des übernächsten Monats

Die Excel-Funktion **MONATSENDE()**

Wofür benötigt man das Monatsende als berechneten Wert?

Auf den ersten Blick erscheint die Funktion wenig hilfreich, da das Datum des letzten eines Monats im Allgemeinen bekannt ist. Aber: Die Funktion lässt sich für weitere Berechnungen oder Darstellungen nutzen, was ihre Bedeutung ausmacht.

So können Sie mit einem benutzerdefinierten Format ermitteln, auf welchen Wochentag das Monatsende des berechneten Datums fällt. Das sehen Sie in der Abbildung oben in Spalte D.

Das entsprechende Datumsformat definieren Sie mit **Strg + 1**. Im **Excel-Tipp zu Datumsformaten** erfahren Sie, wie das funktioniert.

Mit der Funktion **MONATSENDE()** brauchen Sie das Datum des letzten Monats nicht manuell in eine Zelle einzutragen. Sie kopieren einfach die Formel in Zellen, Tabellen oder Listen, in denen Sie diesen Datumswert benötigen. Zum Beispiel für:

- Zeitpläne, Berichte oder Prognosen
- Restlaufzeiten
- Vertragsdauer in Tagen
- Auswertung von Daten mit richtiger Zuordnung zu einem Monat
- Berechnung von Zinsen

So ermitteln Sie das Quartal aus einem Datum

Es gibt in Excel keine Funktion, um zu einem Datumswert das zugehörige Quartal zu berechnen. Dafür braucht es eine Formel. Noch kniffliger wird es, wenn die Zählung des Quartals nicht dem Kalenderjahr, sondern dem Geschäftsjahr entsprechen soll. Wie lauten die entsprechenden Formeln?

Zuletzt geändert am 11.03.2026




Excel verfügt über Funktionen, mit denen Sie aus einem Datumswert wie 02.03.2024 den Tag, den Monat oder das Jahr ermitteln können. Leider gibt es keine Funktion, um das entsprechende Quartal zu ermitteln.

Quartal aus einem Datumswert berechnen

Mit der folgenden Formel können Sie jedoch ohne Probleme die Quartalszahl aus einem Datum ableiten:

=AUFRUNDEN(MONAT(A2)/3; 0)

In der Formel geht man davon aus, dass sich der Datumswert in der Zelle A2 befindet.



The screenshot shows an Excel spreadsheet with a formula bar at the top displaying `=AUFRUNDEN(MONAT(A2)/3;0)`. Below the formula bar is a table with two columns: 'Datum' (Date) and 'Quartal' (Quarter). The table contains 13 rows of data, starting from 01.01.2024 and ending on 01.12.2024. The quarters are calculated based on the dates, with 01.01.2024 being Quarter 1, 01.04.2024 being Quarter 2, and 01.07.2024 being Quarter 3.

	A	B	C	D
1	Datum	Quartal		
2	01.01.2024	1		
3	01.02.2024	1		
4	01.03.2024	1		
5	01.04.2024	2		
6	01.05.2024	2		
7	01.06.2024	2		
8	01.07.2024	3		
9	01.08.2024	3		
10	01.09.2024	3		
11	01.10.2024	4		
12	01.11.2024	4		
13	01.12.2024	4		

Aus dem Datumswert das Quartal berechnen

Quartal aus einem Datumswert ableiten – Geschäftsjahr entspricht nicht Kalenderjahr

Mit der Formel oben wird bei der Quartalsermittlung davon ausgegangen, dass das Geschäftsjahr dem Kalenderjahr entspricht. In manchen Unternehmen beginnt das Geschäftsjahr aber mitten im Kalenderjahr.

Beispiel: Das Geschäftsjahr endet am 30.09. Dann soll die Zählung der Quartale mit dem Beginn des neuen Geschäftsjahres einsetzen. Der 01.10. liegt dann im 1. Quartal, der 01.01.2024 gehört dementsprechend ins 2. Quartal.

Handelt es sich um ein abweichendes Geschäftsjahr, müssen Sie die Formel für die Quartalsermittlung wie folgt anpassen:

=AUFRUNDEN(((MONAT(A2) + (MONAT(A2)<=B2) *12) -B2)/3; 0)

Die Formel geht von den folgenden Annahmen aus:

- **A2:** Zelle, die den Datumswert beinhaltet, für den das Quartal nach Geschäftsjahreszählung ermittelt werden soll
- **B2:** Zelle mit der Monatszahl des Monats, in dem das Geschäftsjahr endet; endet das Geschäftsjahr zum Beispiel am 30.09., dann muss in dieser Zelle der Wert 9 erfasst sein

C2		=AUFRUNDEN(((MONAT(A2)+(MONAT(A2)<=B2)*12)-B2)/3;0)					
	A	B	C	D	E	F	G
1	Datum	Endmonat Geschäftsjahr	Quartal				
2	01.01.2024	9	2				
3	01.01.2024	10	1				
4	01.01.2024	11	1				
5	01.01.2024	12	1				
6	01.01.2024	1	4				
7	01.01.2024	2	4				
8	01.01.2024	3	4				
9	01.01.2024	4	3				
10	01.01.2024	5	3				
11	01.01.2024	6	3				
12	01.01.2024	7	2				
13	01.01.2024	8	2				

Quartal aus Datumswert berechnen für unterschiedliche Geschäftsjahre

Hinweis: Excel-Funktionen für Tag, Monat und Jahr

Mit der Funktion **MONAT()** ermitteln Sie die Monatszahl aus einem Datumswert wie dem 02.03.2024. Das Ergebnis wäre in diesem Fall 3. Entsprechend ermitteln Sie den Tag des Datums (2) mit der Funktion **TAG()** und die Jahreszahl (2024) mit der Funktion **JAHR()**.

Datum in Kalenderwoche umwandeln mit Excel

Wenn Sie die Kalenderwoche für ein Datum ermitteln wollen, müssen Sie beachten, nach welcher Regel sie berechnet wird. Erfahren Sie, mit welchen Formeln Sie die korrekte Kalenderwoche nach dem europäischen System berechnen.

Zuletzt geändert am 11.03.2026




Die Kalenderwoche ist eine wichtige Angabe, die für viele Zwecke verwendet wird. Sie gibt an, in welcher Woche des Jahres sich ein bestimmtes Datum befindet.

In Excel gibt es hierfür eine eigene Funktion mit dem Namen **KALENDERWOCHE()**, mit der Sie die Kalenderwoche aus einem Datum ableiten können. Doch **Vorsicht!** So einfach, wie es aussieht, ist es nicht.

Die zwei Systeme für Kalenderwochen

In der folgenden Abbildung wird mit der Funktion **KALENDERWOCHE()** die Kalenderwoche für den 01.01.2023 berechnet. Als Ergebnis wird die Kalenderwoche 1 ausgegeben.

Was auf den ersten Blick richtig erscheint, ist aber falsch, da der Wert 52 das richtige Ergebnis für dieses Datum ist.



	A	B	C	D
1	01.01.2023	1	=KALENDERWOCHE(A1)	
2				

Kalenderwoche falsch berechnet

Die Funktion **KALENDERWOCHE()** verwendet grundsätzlich zwei Systeme, um die Kalenderwoche zu berechnen:

- **System 1:** Die Woche, die den **1. Januar** enthält, ist die erste Woche des Jahres und die nummerierte Woche 1.
- **System 2:** Die Woche, die den **ersten Donnerstag** des Jahres enthält, ist die erste Kalenderwoche mit der Nummer 1. Dieses System entspricht der in ISO 8601 angegebenen Methode, die im Allgemeinen als europäisches Wochenummerierungssystem bezeichnet wird.

Wenn Sie es nicht explizit in der Funktion **KALENDERWOCHE()** angeben, dann verwendet Excel das **System 1**, das nicht unserem europäischen Wochenummerierungssystem entspricht. Die Formel führt also zu einer „falschen“ Kalenderwoche.

Kalenderwoche nach ISO 8601 berechnen

Dieses Problem lösen Sie, indem Sie das optionale Argument **Zahl_Typ** der Funktion verwenden – also diese Funktion:

=KALENDERWOCHE(Datum; [Zahl_Typ])

Hier geben Sie als **Zahl_Typ** den Wert **21** an.

B1					
	A	B	C	D	E
1	01.01.2023	52	=KALENDERWOCHE(A1;21)		
2					

Die korrekte Kalenderwoche nach ISO 8601

Für diese Problemstellung wurde von Microsoft auch die Funktion **ISOKALENDERWOCHE()** in Excel eingeführt, mit der Sie immer automatisch die **Kalenderwoche nach der europäischen Norm** berechnen können.

B1			  	=ISOKALENDERWOCHE(A1)
	A	B	C	
1	01.01.2023	52	=ISOKALENDERWOCHE(A1)	
2				

Funktion ISOKALENDERWOCHE() in Excel

Aber auch ohne die speziellen Funktionen für die Kalenderwoche können Sie die Kalenderwoche nach der europäischen Norm in Excel berechnen. Mit der folgenden Formel, die sich aus bekannten Standardfunktionen zusammensetzt, berechnen Sie die Kalenderwoche nach der europäischen Norm:

=KÜRZEN((A1-WOCHENTAG(A1; 2) - DATUM(JAHR(A1+4-WOCHENTAG(A1; 2)); 1; -10))/7)

B1		  		=KÜRZEN((A1-WOCHENTAG(A1;2)-DATUM(JAHR(A1+4-WOCHENTAG(A1;2));1;-10))/7)	
	A	B	C	D	E
1	01.01.2023	52	=KÜRZEN((A1-WOCHENTAG(A1;2)-DATUM(JAHR(A1+4-WOCHENTAG(A1;2));1;-10))/7)		
2					
3					

Allgemeine Formel für die Berechnung der Kalenderwoche

Datum eines Wochentags aus Kalenderwoche berechnen

Sie wollen das Datum für den Montag in der 14. Kalenderwoche wissen? Welches Datum mit dieser Angabe gemeint ist, kann man nicht erkennen. Mit der folgenden Formel lässt sich das Datum für einen bestimmten Wochentag schnell ermitteln, wenn Kalenderwoche und Jahr gegeben sind.

Zuletzt geändert am 11.03.2026



„Können wir unsere nächste Projektbesprechung am Montag in der Kalenderwoche 14 durchführen?“

Könnte man schon – aber welches Datum ist das genau? Solche Fragen stellen sich bei der Terminplanung immer wieder.

Für diese Aufgabenstellung gibt es keine Standardfunktion in Excel. Mit der folgenden Formel können Sie für eine beliebige Kalenderwoche eines beliebigen Jahres ermitteln, welches Datum der Montag in der Kalenderwoche hat:

=("4.1."&B1) + B2*7 - 7 - REST("2.1."&B1; 7)

Die Formel geht hierbei von den folgenden Annahmen aus:

- **B1**: Jahreszahl
- **B2**: Kalenderwoche

	A	B	C	D	E	F
1	Jahr	2022				
2	Kalenderwoche	14				
3						
4	Datum	04.04.2022	Montag			
5		05.04.2022	Dienstag			
6						

Datum für den Wochentag Montag einer Kalenderwoche berechnen

Wollen Sie einen anderen Wochentag berechnen, dann fügen Sie am Ende der Formel noch eine entsprechende Addition ein:

- $=("4.1."&B1) + B2*7 - 7 - \text{REST}("2.1."&B1; 7) + \mathbf{1}$ für Dienstag
- $=("4.1."&B1) + B2*7 - 7 - \text{REST}("2.1."&B1; 7) + \mathbf{2}$ für Mittwoch
- $=("4.1."&B1) + B2*7 - 7 - \text{REST}("2.1."&B1; 7) + \mathbf{3}$ für Donnerstag
- $=("4.1."&B1) + B2*7 - 7 - \text{REST}("2.1."&B1; 7) + \mathbf{4}$ für Freitag
- $=("4.1."&B1) + B2*7 - 7 - \text{REST}("2.1."&B1; 7) + \mathbf{5}$ für Samstag
- $=("4.1."&B1) + B2*7 - 7 - \text{REST}("2.1."&B1; 7) + \mathbf{6}$ für Sonntag

Hinweis: Damit das Datum richtig in der Zelle dargestellt wird, müssen Sie der Ergebniszelle noch das **Zahlenformat Datum** zuweisen. Das können Sie entsprechend einstellen, wenn Sie Strg + 1 wählen.

Anzahl der Feiertage zwischen zwei Datumswerten in Excel berechnen

Wie viele Feiertage liegen zwischen einem angegebenen Start- und Enddatum? Mit dieser Formel berechnen Sie die Zahl der Feiertage und nutzen das Ergebnis, um Lieferzeiten oder Projektlaufzeiten zu ermitteln.

Zuletzt geändert am 11.03.2026



Die Berechnung der Anzahl der Feiertage zwischen zwei Datumswerten ist in der betrieblichen Praxis ein wichtiger Bestandteil von Arbeitszeit- und Projektplanungen. Beispielsweise kann es notwendig sein, die tatsächlichen Arbeitstage zu berechnen, um

- die Personaleinsatzplanung zu optimieren,
- Lieferzeiten besser abschätzen zu können oder
- Fristen korrekt zu bestimmen.

Feiertage und freie Tage spielen dabei eine wichtige Rolle, da sie je nach Region und Jahr **variieren**.

Mit der folgenden Kombination aus Funktionen und einer Liste der Feiertage in einer Excel Tabelle berechnen Sie die Anzahl der Feiertage zwischen zwei Datumswerten.

Formel zur Berechnung der Anzahl der Feiertage

Diese Aufgabe können Sie mit der folgenden Formel lösen:

{=SUMME((A2:A21 >=F1) * (A2:A21 <=F2))}

Die einzelnen Bezüge oder Bereiche in der Formel stehen hierbei für die folgenden Parameter:

- **A2:A21:** Feiertagsliste; hier sind die Feiertage im Datumsformat aufgelistet
- **F1:** Startdatum des Zeitintervalls, in dem Sie die Feiertage zählen wollen
- **F2:** Enddatum des Zeitintervalls, in dem Sie die Feiertage zählen wollen

Erfassen Sie die Formel ohne die geschweiften Klammern am Anfang und Ende. Da es sich um eine Matrixformel handelt, bestätigen Sie die Formel mit der Tastenkombination **Strg + Umschalttaste + Enter**. Die geschweiften Klammern werden durch diese Tastenkombination automatisch eingefügt.

F4 {=SUMME((A2:A21>=F1)*(A2:A21<=F2))}					
A	B	C	D	E	F
1 Datum	Feiertag	Gesetzlicher Feiertag in folgenden Bundesländern:		Start	01.01.2025
2 01.01.2025	Neujahr	bundesweit		Ende	10.01.2025
3 06.01.2025	Heilige Drei Könige	Baden-Württemberg, Bayern, Sachsen-Anhalt			
4 08.03.2025	Internationaler Frauentag	Berlin, Mecklenburg-Vorpommern		Anzahl	2
5 18.04.2025	Karfreitag	bundesweit			
6 20.04.2025	Ostersonntag	Brandenburg (Einzig das Land Brandenburg behandelt den Ostersonntag explizit als gesetzlichen Feiertag.)			
7 21.04.2025	Ostermontag	bundesweit			
8 01.05.2025	Tag der Arbeit	bundesweit			
9 29.05.2025	Christi Himmelfahrt	bundesweit			
10 08.06.2025	Pfingstsonntag	Brandenburg (Einzig das Land Brandenburg behandelt den Ostersonntag explizit als gesetzlichen Feiertag.)			
11 09.06.2025	Pfingstmontag	bundesweit			
12 19.06.2025	Fronleichnam	Baden-Württemberg, Bayern, Hessen, Nordrhein-Westfalen, Rheinland-Pfalz, Saarland, In Sachsen nur teilweise. *, In Thüringen nur teilweise. **			
13 08.08.2025	Augsburger Friedensfest	Nur im Stadtgebiet von Augsburg (nicht jedoch im angrenzenden Umland)			
14 15.08.2025	Mariä Himmelfahrt	Saarland, In Bayern nur teilweise. ***			
15 20.09.2025	Weltkindertag	Thüringen			
16 03.10.2025	Tag der deutschen Einheit	bundesweit			
17 31.10.2025	Reformationstag	Brandenburg, Bremen, Hamburg, Mecklenburg-Vorpommern, Niedersachsen, Sachsen, Sachsen-Anhalt, Schleswig-Holstein, Thüringen			
18 01.11.2025	Allerheiligen	Baden-Württemberg, Bayern, Nordrhein-Westfalen, Rheinland-Pfalz, Saarland			
19 19.11.2025	Buß- und Betttag	Sachsen			
20 25.12.2025	1. Weihnachtsfeiertag	bundesweit			
21 26.12.2025	2. Weihnachtsfeiertag	bundesweit			

Formel zur Berechnung der Zahl der Feiertage in einem Zeitraum

Wie ist die Formel aufgebaut?

Mit dem Ausdruck **A2:A21>=F1** wird überprüft, ob die Feiertagswerte im Bereich A2:A21 größer oder gleich dem Startdatum sind. Trifft dies zu, dann wird der Wert **1** (WAHR), sonst (FALSCH) zurückgegeben.

Durch den Formelteil **A2:A21<=F2** wird überprüft, ob die Feiertagswerte kleiner oder gleich dem Enddatum sind. Trifft dies zu, dann wird der Wert **1** (WAHR), sonst (FALSCH) zurückgegeben.

Anschließend werden die Werte der beiden Prüfungen für jeden Feiertag **multipliziert**; dies entspricht der logischen **UND-Prüfung**. Nur bei Feiertagen, die sich innerhalb des Zeitraums befinden, ergibt sich der Wert 1 (1×1). Bei den anderen Feiertagen ergibt sich der Wert 0 (0×1 , 1×0 oder 0×0).

Zum Schluss werden die Ergebnisse der Multiplikationen durch die Funktion **SUMME()** addiert und Sie erhalten die Anzahl der Feiertage, die sich innerhalb des Intervalls befinden.

Alternativ können Sie auch die folgende Formel einsetzen:

=SUMMENPRODUKT((A2:A21 >=F1) * (A2:A21 <=F2))

Die Funktionsweise der Formel ist hier identisch mit der ersten Funktion.

F4 $\sum$ $\times$ $\div$ $\sqrt{x}$ $\sqrt[3]{x}$ $\sqrt[n]{x}$ =SUMMENPRODUKT((A2:A21>=F1)*(A2:A21<=F2))						
	A	B	C	D	E	F
1	Datum	Feiertag	Gesetzlicher Feiertag in folgenden Bundesländern:		Start	01.01.2025
2	01.01.2025	Neujahr	bundesweit		Ende	10.01.2025
3	06.01.2025	Heilige Drei Könige	Baden-Württemberg, Bayern, Sachsen-Anhalt			
4	08.03.2025	Internationaler Frauentag	Berlin, Mecklenburg-Vorpommern		Anzahl	2
5	18.04.2025	Karfreitag	bundesweit			
6	20.04.2025	Ostersonntag	Brandenburg (Einzig das Land Brandenburg behandelt den Ostersonntag explizit als gesetzlichen Feiertag.)			
7	21.04.2025	Ostermontag	bundesweit			
8	01.05.2025	Tag der Arbeit	bundesweit			
9	29.05.2025	Christi Himmelfahrt	bundesweit			
10	08.06.2025	Pfingstsonntag	Brandenburg (Einzig das Land Brandenburg behandelt den Ostersonntag explizit als gesetzlichen Feiertag.)			
11	09.06.2025	Pfingstmontag	bundesweit			
12	19.06.2025	Fronleichnam	Baden-Württemberg, Bayern, Hessen, Nordrhein-Westfalen, Rheinland-Pfalz, Saarland, In Sachsen nur teilweise. *, In Thüringen nur teilweise. **			
13	08.08.2025	Augsburger Friedensfest	Nur im Stadtgebiet von Augsburg (nicht jedoch im angrenzenden Umland)			
14	15.08.2025	Mariä Himmelfahrt	Saarland, In Bayern nur teilweise. ***			
15	20.09.2025	Weltkindertag	Thüringen			
16	03.10.2025	Tag der deutschen Einheit	bundesweit			
17	31.10.2025	Reformationstag	Brandenburg, Bremen, Hamburg, Mecklenburg-Vorpommern, Niedersachsen, Sachsen, Sachsen-Anhalt, Schleswig-Holstein, Thüringen			
18	01.11.2025	Allerheiligen	Baden-Württemberg, Bayern, Nordrhein-Westfalen, Rheinland-Pfalz, Saarland			
19	19.11.2025	Buß- und Bettag	Sachsen			
20	25.12.2025	1. Weihnachtsfeiertag	bundesweit			
21	26.12.2025	2. Weihnachtsfeiertag	bundesweit			

Berechnung der Zahl der Feiertage mit SUMMENPRODUKT()

So können Sie die Nettoarbeitstage in Excel berechnen

Wie viele Arbeitstage liegen zwischen zwei Terminen? Mit Excel können Sie die Nettoarbeitstage mit einer Formel leicht berechnen. Dabei können Sie auch Feiertage und freie Tage berücksichtigen. Und Sie können die arbeitsfreien Tage in der Woche selbst definieren.

Zuletzt geändert am 11.03.2026



Um zu ermitteln, wie viele Arbeitstage in einen Zeitraum fallen, stellt Excel Ihnen die Funktion **NETTOARBEITSTAGE()** zur Verfügung.

Mit dieser Funktion können Sie berechnen, wie viele Arbeitstage zwischen einem Beginn- und Endtermin liegen. Das ist sehr hilfreich, wenn es zum Beispiel um die Planung von Arbeitspaketen im Projektmanagement geht oder um die Auswertung von Arbeitszeiten.

Formel für die Funktion Nettoarbeitstage

Die Funktion besitzt die folgende Syntax:

NETTOARBEITSTAGE(Ausgangsdatum; Enddatum; [Freie_Tage])

- **Ausgangsdatum:** Ein Datum, das das Ausgangs- oder Beginndatum angibt.
- **Enddatum:** Ein Datum, das das Enddatum angibt.
- **Freie_Tage** (optional): Ein optionaler Bereich von einer oder mehreren Datumsangaben, die alle Arten von arbeitsfreien Tagen repräsentieren, die aus dem Arbeitskalender ausgeschlossen werden sollen.

So arbeiten Sie mit der Funktion Nettoarbeitstage

Ein Beispiel soll zeigen, wie Sie die Funktion NETTOARBEITSTAGE nutzen: Es sollen die Arbeitstage zwischen dem 01.01.2022 und dem 09.01.2022 ermittelt werden. Hierbei sollen die Wochenenden (Samstag, Sonntag) nicht berücksichtigt werden. Aus Vereinfachungsgründen wird der Feiertag am 06.01.2022 zunächst nicht berücksichtigt.

Hierzu müssen Sie lediglich das Anfangsdatum (01.01.2022 in Zelle A1) und Enddatum (09.01.2022 in Zelle I1) als Argumente in der Funktion erfassen. Als Ergebnis bekommen Sie den Wert 5 geliefert.

B8									
	A	B	C	D	E	F	G	H	I
1	01.01.2022	02.01.2022	03.01.2022	04.01.2022	05.01.2022	06.01.2022	07.01.2022	08.01.2022	09.01.2022
2	Samstag	Sonntag	Montag	Dienstag	Mittwoch	Donnerstag	Freitag	Samstag	Sonntag
3									
4									
5	von	01.01.2022							
6	bis	09.01.2022							
7									
8	Arbeitstage	5							
9									
10									

Funktion NETTOARBEITSTAGE berechnet die Arbeitstage ohne Samstag und Sonntag

Wie Sie freie Tage bei Nettoarbeitstagen berücksichtigen

Nun sollen auch die jeweiligen Feiertage oder andere freie Tage berücksichtigt werden. Hierzu müssen Sie die Feiertage als optionales Argument entsprechend in der Funktion **NETTOARBEITSTAGE()** angeben.

Bezogen auf das Beispiel referenzieren Sie alle Feiertage des Jahres 2022 als zusätzlichen Bereich E5:E22. In dieser Liste ist das jeweilige Datum für den Feiertag genannt; in Spalte F erkennen Sie, um welchen Feiertag es sich jeweils handelt.

Unter Berücksichtigung der Feiertage erhalten Sie jetzt als Ergebnis den Wert 4 berechnet.

B4						=NETTOARBEITSTAGE(A1;1;E5:E22)			
	A	B	C	D	E	F	G	H	I
1	01.01.2022	02.01.2022	03.01.2022	04.01.2022	05.01.2022	06.01.2022	07.01.2022	08.01.2022	09.01.2022
2	Samstag	Sonntag	Montag	Dienstag	Mittwoch	Donnerstag	Freitag	Samstag	Sonntag
3									
4	Arbeitstage	4			Feiertage				
5					01.01.2022 Neujahr 2022				
6					06.01.2022 Heilige Drei Könige 2022				
7					15.04.2022 Karfreitag 2022				
8					17.04.2022 Ostersonntag 2022				
9					18.04.2022 Ostermontag 2022				
10					01.05.2022 Tag der Arbeit 2022				
11					26.05.2022 Christi Himmelfahrt 2022				
12					05.06.2022 Pfingstsonntag 2022				
13					06.06.2022 Pfingstmontag 2022				
14					16.06.2022 Fronleichnam 2022				
15					15.08.2022 Mariä Himmelfahrt 2022				
16					20.09.2022 Weltkindertag 2022				
17					03.10.2022 Tag der Deutschen Einheit 2022				
18					31.10.2022 Reformationstag 2022				
19					01.11.2022 Allerheiligen 2022				
20					16.11.2022 Buß- und Betttag 2022				
21					25.12.2022 1 Weihnachtstag 2022				
22					26.12.2022 2 Weihnachtstag 2022				
23									

Funktion NETTOARBEITSTAGE mit Feiertagen und freien Tagen

In der Liste der Feiertage im Bereich E5:E22 können beliebige Tage mit Ihrem Datum eingetragen werden; das müssen keine (offiziellen) Feiertage sein.

Möglich ist zum Beispiel, dass Sie Werkschließungen oder Betriebsferien hier erfassen und bei der Berechnung der Nettoarbeitstage dann berücksichtigen.

So können Sie vom Standardwochenende abweichen

Die Funktion NETTOARBEITSTAGE() ignoriert automatisch alle Samstage und Sonntage. Wenn Sie Wochenenden als Arbeitstage berücksichtigen oder die Wochenenden individuell definieren möchten, dann setzen Sie die Funktion **NETTOARBEITSTAGE.INTL()** anstelle von NETTOARBEITSTAGE() ein.

Die Funktion NETTOARBEITSTAGE.INTL() besitzt grundsätzlich den gleichen Aufbau wie NETTOARBEITSTAGE(). Als zusätzliches Argument können Sie hier das Wochenende individuell definieren. NETTOARBEITSTAGE.INTL() besitzt die folgende Syntax:

=NETTOARBEITSTAGE.INTL(Ausgangsdatum; Enddatum; [**Wochenende**];
[Freie_Tage])

Als optionales Argument für **Wochenende** können Sie hier die folgenden Werte erfassen:

- 1 oder nicht angegeben – Samstag, Sonntag
- 2 – Sonntag, Montag
- 3 – Montag, Dienstag
- 4 – Dienstag, Mittwoch
- 5 – Mittwoch, Donnerstag
- 6 – Donnerstag, Freitag
- 7 – Freitag, Samstag
- 11 – nur Sonntag
- 12 – nur Montag
- 13 – nur Dienstag
- 14 – nur Mittwoch
- 15 – nur Donnerstag
- 16 – nur Freitag
- 17 – nur Samstag

Hinweis: Die Angabe des Arguments Wochenende in NETTOARBEITSTAGE.INTL() ist optional. Für den Fall, dass die Angabe fehlt oder 1 entspricht, erhalten Sie ein identisches Ergebnis wie mit der Funktion NETTOARBEITSTAGE().

Für das Beispiel sollen nun Sonntag und Montag nicht als Arbeitstag berücksichtigt werden. Erfassen Sie daher als Wochenend-Argument den Wert 2. Als Ergebnis erhalten Sie den Wert 4; Dienstag, Mittwoch, Freitag und Samstag sind nun die Nettoarbeitstage.

So können Sie die Nettoarbeitstage in Excel berechnen

B4									
	A	B	C	D	E	F	G	H	I
1	01.01.2022	02.01.2022	03.01.2022	04.01.2022	05.01.2022	06.01.2022	07.01.2022	08.01.2022	09.01.2022
2	Samstag	Sonntag	Montag	Dienstag	Mittwoch	Donnerstag	Freitag	Samstag	Sonntag
3									
4	Arbeitstage	4			Feiertage				
5					01.01.2022 Neujahr 2022				
6					06.01.2022 Heilige Drei Könige 2022				
7					15.04.2022 Karfreitag 2022				
8					17.04.2022 Ostersonntag 2022				
9					18.04.2022 Ostermontag 2022				
10					01.05.2022 Tag der Arbeit 2022				
11					26.05.2022 Christi Himmelfahrt 2022				
12					05.06.2022 Pfingstsonntag 2022				
13					06.06.2022 Pfingstmontag 2022				
14					16.06.2022 Fronleichnam 2022				
15					15.08.2022 Mariä Himmelfahrt 2022				
16					20.09.2022 Weltkindertag 2022				
17					03.10.2022 Tag der Deutschen Einheit 2022				
18					31.10.2022 Reformationstag 2022				
19					01.11.2022 Allerheiligen 2022				
20					16.11.2022 Buß- und Bettag 2022				
21					25.12.2022 1 Weihnachtstag 2022				
22					26.12.2022 2 Weihnachtstag 2022				
23									

Funktion NETTOARBEITSTAGE.INTL mit individuellem Wochenende, Feiertagen und freien Tagen

Mit ARBEITSTAG.INTL-Funktion ein Zieldatum berechnen

Wie Sie mit der Excel-Funktion ARBEITSTAG.INTL ein Ende-Datum berechnen und so zuverlässige Terminzusagen machen. Dabei berücksichtigen Sie Wochenende und freie Tage, die Sie flexibel festlegen können.

Zuletzt geändert am 11.03.2026



Was leistet die Excel-Funktion ARBEITSTAG.INTL()

Mit der Excel-Funktion **ARBEITSTAG.INTL()** können Sie ein **Zieldatum berechnen**, basierend auf einem Ausgangsdatum und einer bestimmten Anzahl von Arbeitstagen.

Das Zieldatum kann sowohl in der Zukunft als auch in der Vergangenheit liegen – je nachdem, ob eine positive oder negative Anzahl von Arbeitstagen angegeben wird.

Im Unterschied zur ARBEITSTAG-Funktion erlaubt ARBEITSTAG.INTL die individuelle Festlegung von Wochenend- oder arbeitsfreien Tagen.

Mit dieser Anpassungsmöglichkeit berechnen Sie beispielsweise ein Fälligkeitsdatum, Liefertermine und Fertigstellungsfristen oder Sie ermitteln ein Startdatum auf Grundlage einer vorgegebenen Zeitspanne.

Bei der Berechnung mit der ARBEITSTAG.INTL-Funktion werden ausschließlich Werktage berücksichtigt, wobei optional auch arbeitsfreie Feiertage ausgeschlossen werden können.

Die Parameter der Funktion ARBEITSTAG.INTL()

Die Funktion **ARBEITSTAG.INTL()** besitzt die folgende Syntax:

=ARBEITSTAG.INTL(Ausgangsdatum; Tage; [Wochenende]; [Freie_Tage])

Dabei gilt:

- **Ausgangsdatum:** Das Datum, von dem aus gestartet wird.
- **Tage:** Die Anzahl der Arbeitstage vor oder nach dem Ausgangsdatum.
- **Wochenende** (optional): Eine Angabe, welche Wochentage bei der Berechnung als Wochenenden betrachtet werden sollen.
- **Freie_Tage** (optional): Eine Liste von Daten, die als arbeitsfreie Tage oder Feiertage berücksichtigt werden sollen.

Beispiel für die Funktion ARBEITSTAG.INTL()

Hier ein **Beispiel:** Ausgehend vom 01.01.2024 soll das Enddatum berechnet werden, wenn sieben Arbeitstage für eine Aufgabe, ein Arbeitspaket oder einen Kundenauftrag benötigt werden.

Ausgangsdatum und Arbeitstage sind in der folgenden Abbildung in den Zellen A7 (01.01.2024) und B7 (7) eingetragen. In der Zelle C7 steht das Ergebnis (10.01.2024).

Die Funktion berücksichtigt also, dass der 06.01.2024 und der 07.01.2024 auf ein Wochenende fallen, das bei der Berechnung nicht berücksichtigt wird (arbeitsfrei).

C7												
=ARBEITSTAG.INTL(A7;B7)												
	A	B	C	D	E	F	G	H	I	J	K	L
1	01.01.2024	02.01.2024	03.01.2024	04.01.2024	05.01.2024	06.01.2024	07.01.2024	08.01.2024	09.01.2024	10.01.2024	11.01.2024	12.01.2024
2	Montag	Dienstag	Mittwoch	Donnerstag	Freitag	Samstag	Sonntag	Montag	Dienstag	Mittwoch	Donnerstag	Freitag
3												
4												
5												
6	Ausgangsdatum	Arbeitstage	Enddatum									
7	01.01.2024	7	10.01.2024									
8												
9			=ARBEITSTAG.INTL(A7;B7)									
10												

Wie die Excel-Funktion ARBEITSTAG.INTL() rechnet

Beachten Sie, dass Sie das **Zahlenformat** für die Formelzelle (C7) mit dem Format **DATUM** versehen, damit Sie das Ergebnis der Funktion als Datum angezeigt bekommen.

Wenn Sie das Argument Wochenende nicht festlegen, dann geht die Funktion automatisch davon aus, dass Samstag und Sonntag als Wochenende gelten.

Varianten für die Wochenendtage

Nun soll nur der Sonntag als Wochenende gezählt werden. Das Ausgangsdatum soll der 01.01.2024 bleiben und es soll das Zieldatum nach sechs Arbeitstagen berechnet werden.

Wochenende mit festgelegtem Zahlencode definieren

Hier können Sie jetzt das Argument **Wochenende** nutzen. Für das Argument **Wochenende** können Sie einen Code von 1 bis 17 erfassen. Die Codes haben hierbei die folgende Bedeutung:

- 1 oder nicht angegeben – Samstag, Sonntag
- 2 – Sonntag, Montag
- 3 – Montag, Dienstag
- 4 – Dienstag, Mittwoch
- 5 – Mittwoch, Donnerstag
- 6 – Donnerstag, Freitag
- 7 – Freitag, Samstag
- 11 – nur Sonntag
- 12 – nur Montag
- 13 – nur Dienstag
- 14 – nur Mittwoch
- 15 – nur Donnerstag
- 16 – nur Freitag
- 17 – nur Samstag

Erfassen Sie für Ihre gewünschte Berechnung des Enddatums für das Wochenende (Sonntag) den Zahlencode 11 im Argument **Wochenende** (siehe folgende Abbildung).

C7 =ARBEITSTAG.INTL(A7;B7;11)												
	A	B	C	D	E	F	G	H	I	J	K	L
1	01.01.2024	02.01.2024	03.01.2024	04.01.2024	05.01.2024	06.01.2024	07.01.2024	08.01.2024	09.01.2024	10.01.2024	11.01.2024	12.01.2024
2	Montag	Dienstag	Mittwoch	Donnerstag	Freitag	Samstag	Sonntag	Montag	Dienstag	Mittwoch	Donnerstag	Freitag
3												
4												
5												
6	Ausgangsdatum	Arbeits-tage	Enddatum									
7	01.01.2024	6	08.01.2024									
8												
9			=ARBEITSTAG.INTL(A7;B7;11)									
10												

Ergebnis der Funktion ARBEITSTAG.INTL() für den einen Wochenendtag Sonntag

Freie Eingabe der „Wochenendtage“

Sie können für das Argument **Wochenende** auch Zeichenfolgewerte **aus sieben Zeichen** angeben und so beliebige Tage als „Wochenende“ definieren. Jedes Zeichen steht hierbei für einen Wochentag, beginnend mit Montag.

In der Ziffernfolge für die Arbeits- und Wochenendtage bezeichnet

- **1** einen arbeitsfreien Tag,
- hingegen einen Arbeitstag.

Es sind nur die Zeichen 1 und 0 zulässig. Die Ziffernfolge **0000011** ergibt zum Beispiel ein Wochenende für Samstag und Sonntag. Die Ziffernfolge **1111111** ergibt eine Fehlermeldung, da durch diese Eingabe keine Berechnung möglich ist, da an allen Tagen Wochenende ist.

Wichtig: Erfassen Sie die Ziffernfolge in **Anführungszeichen** in der Funktion; also als **"0010100"**.

Wandeln Sie das vorherige Beispiel ab, indem die freien Wochenendtage immer am Mittwoch und Freitag sein sollen. Für diese Aufgabenstellung können Sie die folgende Ziffernfolge **0010100** als Argument für das Wochenende einsetzen.

Diese Variante ist dann hilfreich, wenn Sie unterschiedliche Schichtpläne oder Arbeitspläne für Ihre Mitarbeitenden beachten wollen.

C7 : X ✓ f_x =ARBEITSTAG.INTL(A7;B7;"0010100")												
1	A	B	C	D	E	F	G	H	I	J	K	L
1	01.01.2024	02.01.2024	03.01.2024	04.01.2024	05.01.2024	06.01.2024	07.01.2024	08.01.2024	09.01.2024	10.01.2024	11.01.2024	12.01.2024
2	Montag	Dienstag	Mittwoch	Donnerstag	Freitag	Samstag	Sonntag	Montag	Dienstag	Mittwoch	Donnerstag	Freitag
3												
4												
5												
6	Ausgangsdatum	Arbeitstage	Enddatum									
7	01.01.2024	6	09.01.2024									
8												
9			=ARBEITSTAG.INTL(A7;B7;"0010100")									
10												

Wochenende flexibel definieren mit 0 und 1

Feiertage und freie Tage in der Berechnung berücksichtigen

Nun wollen Sie noch die Feiertage oder andere freie Tage berücksichtigen. Diese Tage sind im Bereich G7:G8 eingetragen; hier der 01.01.2024 und der 06.01.2024.

Geben Sie diesen Bereich jetzt als Argument **Feiertage** in der Funktion ARBEITSTAG.INTL() ein.

C7												
=ARBEITSTAG.INTL(A7;B7;11;G7:G8)												
	A	B	C	D	E	F	G	H	I	J	K	L
1	01.01.2024	02.01.2024	03.01.2024	04.01.2024	05.01.2024	06.01.2024	07.01.2024	08.01.2024	09.01.2024	10.01.2024	11.01.2024	12.01.2024
2	Montag	Dienstag	Mittwoch	Donnerstag	Freitag	Samstag	Sonntag	Montag	Dienstag	Mittwoch	Donnerstag	Freitag
3												
4												
5												
6	Ausgangsdatum	Arbeitstage	Enddatum				Feiertage					
7	01.01.2024	6	09.01.2024				01.01.2024					
8							06.01.2024					
9			=ARBEITSTAG.INTL(A7;B7;11;G7:G8)									
10												

Feiertage und freie Tage in der Funktion ARBEITSTAG.INTL

Hinweis: Da der 01.01.2024 das Startdatum darstellt, hat dieser Feiertag keine Auswirkung auf das Endergebnis. Der 06.01.2024 hingegen wird als Feiertag berücksichtigt. Er wird nicht als Wochenende (Samstag) berücksichtigt, da als Zahlencode fürs Wochenende 11 eingetragen ist; nur der Sonntag zählt also als Wochenende.

Liste mit Arbeitstagen automatisch erstellen

Sie benötigen eine Liste aller Arbeitstage und wollen Wochenenden und Feiertage aus der Liste ausschließen? Mit dieser Formel erstellen Sie automatisch eine solche Datumsliste.

Zuletzt geändert am 11.03.2026



Im Arbeitsalltag ist es oft notwendig, Zeitpläne zu erstellen, die nur Arbeitstage umfassen und dabei Wochenenden und Feiertage automatisch ausschließen.

Ob für Projektpläne, Abwesenheitslisten oder andere zeitbasierte Aufgaben – Excel bietet eine praktische Lösung, um eine solche Liste von Arbeitstagen zu erstellen – ohne Wochenende und Feiertage.

Datumsliste mit Arbeitstagen erstellen

Mit der folgenden Formel können Sie automatisch eine Liste von Arbeitstagen erstellen:

=ARBEITSTAG.INTL(Ausgangsdatum-1; SEQUENZ(Tage); [Wochenende]; [Freie_Tage])

Dabei gilt:

- **Ausgangsdatum**: Das Datum, von dem aus gestartet wird.
- **Tage**: Die Anzahl der Arbeitstage, die in der Liste erscheinen sollen.
- **Wochenende** (optional): Eine Angabe, welche Wochentage beim Erstellen der Liste als Wochenenden betrachtet werden sollen.
- **Freie_Tage** (optional): Eine Liste von Daten, die als arbeitsfreie Tage oder Feiertage berücksichtigt werden sollen.

Der Parameter **Wochenende** kann folgende Bedeutung haben:

- **1** = Samstag, Sonntag
- **2** = Sonntag, Montag
- **3** = Montag, Dienstag
- **4** = Dienstag, Mittwoch
- **5** = Mittwoch, Donnerstag
- **6** = Donnerstag, Freitag
- **7** = Freitag, Samstag
- **11** = Nur Sonntag
- **12** = Nur Montag
- **13** = Nur Dienstag
- **14** = Nur Mittwoch
- **15** = Nur Donnerstag
- **16** = Nur Freitag
- **17** = Nur Samstag

In der folgenden Abbildung sehen Sie ein Beispiel für eine solche Datumsliste.

C2		=ARBEITSTAG.INTL(A2-1;SEQUENZ(A5);1;A8:A11)				
	A	B	C	D	E	
1	Startdatum		Arbeitstage			
2	01.01.2025		02.01.2025			
3			03.01.2025			
4	Anzahl Tage		07.01.2025			
5	10		08.01.2025			
6			09.01.2025			
7	Feiertage		10.01.2025			
8	01.01.2025		13.01.2025			
9	06.01.2025		14.01.2025			
10	18.04.2025		15.01.2025			
11	21.04.2025		16.01.2025			

Datumsliste für Arbeitstage erstellen

Wie die Datumsliste erzeugt wird

Die Formel berechnet automatisch **ein** Datum, das eine bestimmte Anzahl von Arbeitstagen vor oder nach einem Startdatum liegt. Wochenenden (Samstag und Sonntag) werden dabei ausgeschlossen. Um auch Feiertage auszuschließen, können diese optional als Argument hinzugefügt werden.

Der Trick in dieser Formel beruht darauf, dass Sie die Funktion **SEQUENZ()** in die Funktion **ARBEITSTAG.INTL()** verschachteln und als Inhalt für das Argument **Tage** verwenden.

=ARBEITSTAG.INTL(A2-1; **SEQUENZ(A5)**; 1; A8:A11).

Hierdurch wird im Beispiel eine Liste von Werten erzeugt:

1, 2, 3, 4, 5, 6, 7, 8, 9 und 10

Die Werte werden einzeln als Argument **Tage** an die Funktion ARBEITSTAG.INTL übergeben. Somit entsteht automatisch die entsprechende Auflistung der Arbeitstage.

Arbeitstage für bestimmten Monat ausrechnen

Wie viele Arbeitstage hat ein bestimmter Monat, wenn Wochenenden, freie Tage und Feiertage berücksichtigt werden sollen? Mit der Funktion `NETTOARBEITSTAGE()` und den richtigen Parametern, berechnen Sie diese Zahl genau.

Zuletzt geändert am 11.03.2026



Eine präzise Planung und Verwaltung von Ressourcen sind unerlässlich. Egal, ob in der Buchhaltung, im Personalwesen oder im Projektmanagement – die genaue Ermittlung der Arbeitstage in einem bestimmten Zeitraum spielt eine wichtige Rolle.

Excel bietet eine Vielzahl von Funktionen, die es ermöglichen, solche Berechnungen schnell und einfach durchzuführen.

In diesem Beitrag erfahren Sie, wie Sie mit Excel die **Anzahl der Arbeitstage in einem bestimmten Monat** berechnen, um Ihre Planungsprozesse zu optimieren und Personalressourcen besser zu verwalten.

Beispiel zur Berechnung der Arbeitstage pro Monat

In der folgenden Abbildung sehen Sie in Spalte A eine Auflistung von Monaten, für die rechts daneben in Spalte B die Arbeitstage ermittelt werden sollen.

Beachten Sie, dass der Monat als **Zahl** (Datum) erfasst ist und über das **Zahlenformat** als Monat mit Jahresangabe dargestellt wird. Das entspricht dem benutzerdefinierten **Zahlenformat MMM JJ**.

A2					01.01.2024	
	A	B	C	D	E	F
1	Monat	Arbeitstage			Feiertag	Datum 2024
2	Jan 24				Neujahr	01.01.2024
3	Feb 24				Heilige Drei Könige	06.01.2024
4	Mrz 24				Internationaler Frauentag	08.03.2024
5	Apr 24				Karfreitag	29.03.2024
6	Mai 24				Ostersonntag	31.03.2024
7	Jun 24				Ostermontag	01.04.2024
8	Jul 24				Tag der Arbeit	01.05.2024
9	Aug 24				Christi Himmelfahrt	09.05.2024
10	Sep 24				Pfingstsonntag	19.05.2024
11	Okt 24				Pfingstmontag	20.05.2024
12	Nov 24				Fronleichnam	30.05.2024
13	Dez 24				Augsburger Friedensfest	08.08.2024
14					Mariä Himmelfahrt	15.08.2024
15					Weltkindertag	20.09.2024
16					Tag der deutschen Einheit	03.10.2024
17					Reformationstag	31.10.2024
18					Allerheiligen	01.11.2024
19					Buß- und Betttag	20.11.2024
20					1. Weihnachtsfeiertag	25.12.2024
21					2. Weihnachtsfeiertag	26.12.2024
--						

Liste mit Angaben zur Berechnung der Arbeitstage pro Monat

In der rechten Tabelle (Spalten E und F) sind die Feiertage und freien Tage für das Jahr 2024 aufgelistet, die Sie für die Berechnung der Arbeitstage berücksichtigen wollen.

Formel zur Berechnung der Arbeitstage pro Monat

Sie können die Aufgabenstellung mit der folgenden Formel lösen:

=NETTOARBEITSTAGE(DATUM(JAHR(A2); MONAT(A2); 1); MONATSENDE(A2; 0); \$F\$2:\$F\$21)

- **A2** steht für die Zelle, in welcher der Monat als Zahl erfasst ist.
- **\$F\$2:\$F\$21** steht für die Liste, in welcher die Feiertage aufgelistet sind. Sie sollten die Liste mit **absoluten Bezügen (mit dem Zeichen \$)** erfassen,

damit Sie die Formel ohne Probleme nach unten kopieren können.

Je nach Monatsangabe in der Spalte A wird mit dieser Formel automatisch die richtige Anzahl von Arbeitstagen je Monat berechnet.

B2 : ✕ ✓ <i>f_x</i> =NETTOARBEITSTAGE(DATUM(JAHR(A2);MONAT(A2);1);MONATSENDE(A2;0);\$F\$2:\$F\$21)							
	A	B	C	D	E	F	G
1	Monat	Arbeitstage			Feiertag	Datum 2024	
2	Jan 24	22,00			Neujahr	01.01.2024	
3	Feb 24	21,00			Heilige Drei Könige	06.01.2024	
4	Mrz 24	19,00			Internationaler Frauentag	08.03.2024	
5	Apr 24	21,00			Karfreitag	29.03.2024	
6	Mai 24	19,00			Ostersonntag	31.03.2024	
7	Jun 24	20,00			Ostermontag	01.04.2024	
8	Jul 24	23,00			Tag der Arbeit	01.05.2024	
9	Aug 24	20,00			Christi Himmelfahrt	09.05.2024	
10	Sep 24	20,00			Pfingstsonntag	19.05.2024	
11	Okt 24	21,00			Pfingstmontag	20.05.2024	
12	Nov 24	19,00			Fronleichnam	30.05.2024	
13	Dez 24	20,00			Augsburger Friedensfest	08.08.2024	
14					Mariä Himmelfahrt	15.08.2024	
15					Weltkindertag	20.09.2024	
16					Tag der deutschen Einheit	03.10.2024	
17					Reformationstag	31.10.2024	
18					Allerheiligen	01.11.2024	
19					Buß- und Betttag	20.11.2024	
20					1. Weihnachtsfeiertag	25.12.2024	
21					2. Weihnachtsfeiertag	26.12.2024	

Liste mit den genauen Arbeitstagen pro Monat

Was berechnet die Funktion NETTOARBEITSTAGE()?

Grundsätzlich können Sie die Arbeitstage mit der Funktion **NETTOARBEITSTAGE()** berechnen. NETTOARBEITSTAGE() benötigt die folgenden Argumente:

- Ausgangsdatum
- Enddatum
- Freie Tage

Ausgangsdatum

Das Ausgangsdatum, den 1. eines Monats, leiten Sie in der Formel mit der Funktion **DATUM(JAHR(A2); MONAT(A2); 1)** ab:

```
=NETTOARBEITSTAGE(DATUM(JAHR(A2); MONAT(A2); 1); MONATSENDE(A2;0);  
$F$2:$F$21)
```

In der Zelle A2 befindet sich das Datum (01.01.2024), das aber im benutzerdefinierten Zahlenformat als Monat mit Jahresangabe dargestellt wird (Jan 24).

Standardmäßig wird bei der Datenerfassung immer der Monatserste als Datum verwendet (01.01.2024 für die Zelle A2). Darauf können Sie sich jedoch nicht verlassen, da der 02.01.2024 oder der 03.01.2024 im gewählten Zahlenformat ebenfalls als Jan 24 dargestellt werden würde.

Sie müssen daher als Datum für die Formel unbedingt den Ersten des Monats im Argument Ausgangsdatum erzwingen. Dies erreichen Sie mit der Formel:

DATUM(JAHR(A2); MONAT(A2); 1).

Mit der Funktion DATUM(Jahreszahl; Monatszahl; Tageszahl) können Sie eine Zahl für ein bestimmtes Datum erzeugen. Sie müssen der Funktion DATUM() die folgenden Parameter übergeben:

- **Jahreszahl:** Diese Zahl wird berechnet, indem die Funktion JAHR() auf die Zelle A2 angewendet wird. Die Jahreszahl aus dem Datum 01.01.**2024** beträgt 2024.
- **Monatszahl:** Analog zur Jahreszahl berechnen Sie diesen Parameter mit der Funktion **MONAT()**, die ebenfalls auf das Datum in Zelle A2 angewendet wird. Die Monatszahl für das Datum 01.**01**.2024 beträgt 1.
- **Tageszahl:** Da immer zwingend der Monatserste verwendet werden soll, erfassen Sie hier die Zahl 1 als fixen Wert.

Aufgrund dieser Berechnung wird für die Zelle A2 das Datum 01.01.2024 als Ausgangsdatum als Parameter für die Funktion NETTOARBEITSTAGE() übergeben.

Enddatum

Das Enddatum für den Monat können Sie mit der Funktion **MONATSENDE()** ermitteln. Mit dieser Funktion wird automatisch das letzte Datum für einen bestimmten Monat berechnet:

```
=NETTOARBEITSTAGE(DATUM(JAHR(A2); MONAT(A2); 1); MONATSENDE(A2; 0);  
$F$2:$F$21)
```

Im ersten Argument (**Ausgangsdatum**) beziehen Sie sich auf die Zelle A2, in der sich der entsprechende Monat aus dem eingetragenen Datum ergibt, für den Sie den Monatsletzten berechnen wollen.

Im zweiten Argument (**Monate**) geben Sie die Anzahl der Monate an, die addiert (positive Zahl) oder subtrahiert (negative Zahl) werden sollen. Da der letzte Tag des angegebenen Monats berechnet werden soll, setzen Sie diese Zahl auf 0.

Freie Tage

Das optionale Argument Freie Tage verweist auf die Liste (Bezug) mit den freien Tagen oder **Feiertagen**. Im Beispiel stehen diese im Bereich **\$F\$2:\$F\$21**.

```
=NETTOARBEITSTAGE(DATUM(JAHR(A2); MONAT(A2); 1); MONATSENDE(A2; 0);  
$F$2:$F$21)
```

Excel berücksichtigt diese Datumswerte bei der Berechnung der Arbeitstage automatisch. Fällt ein Feiertag auf einen normalen Arbeitstag, dann wird dieser Tag nicht bei der Berechnung der Arbeitstage berücksichtigt.

Wie Sie Schaltjahre mit Excel berechnen

Nicht jede durch 4 teilbare Jahreszahl ist gleichbedeutend mit einem Schaltjahr. Es gibt Ausnahmen. Mit einer einfachen Abfrage ermitteln Sie mit Excel, ob ein bestimmtes Jahr ein Schaltjahr ist.

Zuletzt geändert am 11.03.2026



Nicht jedes Kalenderjahr hat gleich viele Tage. Regelmäßig gibt es Schaltjahre, die einen zusätzlichen Tag aufweisen.

In Schaltjahren wird der Februar um einen Schalttag verlängert. Dadurch erhalten diese Jahre einen 29. Februar, der dann der 60. Tag des gregorianischen Kalenders ist. Somit bleiben 306 Tage bis zum Jahresende.

Schaltjahre können eine Kalkulation oder Planung schnell durcheinanderbringen. Dann ist es wichtig zu wissen, ob ein bestimmtes Jahr ein Schaltjahr ist – oder nicht.

Woraus ergibt sich ein Schaltjahr?

Um Schaltjahre berechnen zu können, müssen Sie grundsätzlich die gregorianische Schalttagesregelung berücksichtigen.

Die gregorianische Schalttagesregelung besteht aus den folgenden drei einzelnen Regeln:

- Die **durch 4 ganzzahlig teilbaren Jahre** sind, abgesehen von den folgenden Ausnahmen, Schaltjahre.
- **Säkularjahre**, also die Jahre, die ein Jahrhundert abschließen (zum Beispiel 1800, 1900, 2100 und 2200), sind, abgesehen von der folgenden Ausnahme, **keine Schaltjahre**.
- Die **durch 400 ganzzahlig teilbaren Säkularjahre**, zum Beispiel das Jahr 2000, sind **Schaltjahre**.

Schaltjahr mit Excel berechnen

Da Excel über ein eigenes Datumssystem verfügt, können Sie sich dieses zunutze machen, ohne dass Sie mathematisch komplizierte Formeln entwickeln müssen, um ein Schaltjahr zu berechnen.

Mit der folgenden Formel können Sie überprüfen, ob eine Jahreszahl ein Schaltjahr ist:

=WENN(ISTFEHLER(("29.02."&A1)*1); "kein Schaltjahr"; "Schaltjahr")

Die Formel geht davon aus, dass sich die zu überprüfende Jahreszahl in der Zelle A1 befindet.

The screenshot shows an Excel spreadsheet with the following content:

	A	B	C	D	E	F	G	H
1	2024							
2								
3								
4	Schaltjahr							
5								

The formula bar at the top shows the formula in cell A4: `=WENN(ISTFEHLER(("29.02."&A1)*1); "kein Schaltjahr"; "Schaltjahr")`. The formula is highlighted with a red box.

Formel für das Erkennen von Schaltjahren

So funktioniert die Formel:

- Als Erstes wird mit der Jahreszahl in der Zelle A1 eine Textfolge erstellt (für das Jahr 2024 ist dies: 29.02.2024):
`=WENN(ISTFEHLER(("29.02."&A1)*1); "kein Schaltjahr"; "Schaltjahr")`
- Die Textfolge wird mit 1 multipliziert:
`=WENN(ISTFEHLER(("29.02."&A1)*1); "kein Schaltjahr"; "Schaltjahr")`
- Durch die Multiplikation mit 1 wird die Textfolge in eine Zahl (Datumswert) umgewandelt.
- Ist der 29.02.2024 (das gewählte Beispiel) im Datumssystem vorhanden, dann wandelt Excel die Textfolge 29.02.2024 in die entsprechende Zahl um.
- Ist der 29.02.2024 nicht im Datumssystem vorhanden, dann ergibt die Multiplikation den Fehlerwert **#WERT!**
- Das Ergebnis der Multiplikation wird durch die Funktion **ISTFEHLER()** auf einen Fehler untersucht:
`=WENN(ISTFEHLER(("29.02."&A1)*1); "kein Schaltjahr"; "Schaltjahr")`

- Bei einem Fehler #WERT! wird der Wahrheitswert **WAHR** als Ergebnis durch die Funktion ISTFEHLER() übermittelt.
- Die Bedingung der WENN-Funktion ist somit erfüllt, und es wird der Text **kein Schaltjahr** als Ergebnis ausgegeben:
`=WENN(ISTFEHLER(("29.02."&A1)*1); "kein Schaltjahr"; "Schaltjahr")`
- Bei keinem Fehler wird der Wahrheitswert **FALSCH** als Ergebnis durch die Funktion ISTFEHLER() übermittelt.
- Die Bedingung der WENN-Funktion ist somit nicht erfüllt und es wird der Text **Schaltjahr** als Ergebnis ausgegeben:
`=WENN(ISTFEHLER(("29.02."&A1)*1); "kein Schaltjahr"; "Schaltjahr")`

Anstelle der Textausgabe können auch andere Ausgaben mit der Formel erzeugt werden. Wenn Sie beispielsweise für Berechnungen wissen wollen, ob das betrachtete Jahr 365 oder 366 Tage hat, nutzen Sie die Formel:

=WENN(ISTFEHLER(("29.02."&D1)*1); 365; 366)

Hinweis

Excel und das falsche Schaltjahr 1900

Excel – und damit auch die hier vorgestellte Formel – geht fälschlicherweise davon aus, dass das Jahr 1900 ein Schaltjahr ist. Der Fehler besteht schon von Beginn an und wurde bislang nicht korrigiert, weil die Fehlerbeseitigung erhebliche Nachteile für Excel-Nutzer hätte.

Alle Formeln, die Datumsangaben nutzen, müssten korrigiert werden. Ein riesiger Aufwand für Excel-Nutzer. Der Fehler wirkt sich auch nur auf ein Datum aus, das vor dem 1.3.1900 liegt. Diese Tage dürften für die meisten Excel-Nutzer keine Rolle spielen.

Fazit: Die Berechnung der Schaltjahre mit der angegebenen Formel funktioniert erst für Jahre nach 1900.

Umstellungsdatum für Sommer- und Winterzeit berechnen

Mit einer einfachen Excel-Formel berechnen Sie das Datum, an dem von Winterzeit auf Sommerzeit umgestellt wird. Entsprechend funktioniert auch die Berechnung des Datums für den Wechsel von Sommerzeit auf Winterzeit.

Zuletzt geändert am 11.03.2026



Die **Sommerzeit** (Daylight Saving Time) stellt ab März in den Sommermonaten die Uhren um eine Stunde vor, um das Tageslicht optimal zu nutzen und Energie zu sparen.

Die **Winterzeit**, auch Normalzeit genannt, beginnt dann am letzten Sonntag im Oktober, wenn wieder zur standardisierten Zeit zurückgekehrt wird.

Das jeweilige Umstellungsdatum für Sommer- und Winterzeit können Sie ohne große Probleme mit Excel berechnen.

Datum der Umstellung auf Sommerzeit berechnen

Die Sommerzeit beginnt immer am letzten Sonntag im März eines Jahres. Zu Beginn der Sommerzeit wird die Uhrzeit um 02:00 Uhr um eine Stunde auf 03:00 Uhr vorgestellt. Mit der folgenden Formel können Sie das Umstellungsdatum auf die Sommerzeit in Excel berechnen:

=DATUM(A1; 4;) - WOCHENTAG(DATUM(A1; 4;)) + 1

Die Formel geht davon aus, dass sich die entsprechende **Jahreszahl**, für welche das Umstellungsdatum der Sommerzeit berechnet werden soll, in der **Zelle A1** befindet.

A4					
	A	B	C	D	E
1	2024				
2					
3					
4	31.03.2024				
5					

Formel zur Berechnung des Beginns der Sommerzeit

Umstellungsdatum der Winterzeit berechnen

Die Winterzeit beginnt am letzten Sonntag im Oktober eines Jahres. Zu Beginn der Winterzeit wird die Uhrzeit um 03:00 Uhr um eine Stunde auf 02:00 Uhr zurückgestellt.

Mit der folgenden Formel können Sie das Umstellungsdatum auf die Winterzeit in Excel berechnen:

=DATUM(A1; 11;) - WOCHENTAG(DATUM(A1; 11;)) + 1

A4					
	A	B	C	D	E
1	2024				
2					
3					
4	27.10.2024				

Formel zur Berechnung des Beginns der Winterzeit

Hinweis: Falls das Ergebnis nicht als Datum angezeigt wird, dann müssen Sie der Zelle noch ein Datumsformat zuweisen. Markieren Sie hierzu die Formelzelle und drücken die Tastenkombination **Strg + 1**. Hier können Sie unter dem Register **Zahlen** ein Datumsformat auswählen.

Mit Datumsangaben vor 1900 in Microsoft Excel rechnen

Für Excel ist der 1.1.1900 der Tag 1 der Zeitrechnung. Wer mit einem Datum davor arbeiten und rechnen will, muss ein paar Umwege gehen. Lesen Sie, wie Sie ältere Datumsangaben auswerten und was der Trick mit dem Gregorianischen Kalender damit zu tun hat.

Zuletzt geändert am 11.03.2026



Die Arbeit mit Datumsangaben in Microsoft Excel kann herausfordernd sein, insbesondere wenn Sie mit **Datumswerten vor dem Jahr 1900** rechnen müssen.

Excel verwendet das **Datumssystem ab dem 1.1.1900**. Jegliche Datumsangaben davor werden als Text behandelt und können nicht direkt in Berechnungen einbezogen werden.

Wenn Sie beispielsweise das Datum 31.12.1899 eingeben, wird es nicht als Datum erkannt und bleibt linksbündig im **Zellenformat Text** stehen.

	A	B
1	01.01.1900	
2	31.12.1899	
3		
4		

Wie Excel mit Datumseingaben umgeht

Standardmäßig erkennt Excel Datumsangaben erst ab dem 1. Januar 1900. Aber es gibt dennoch eine Möglichkeit, mit früheren Datumswerten zu rechnen und zu arbeiten.

Geheimtrick Gregorianischer Kalender

Der **Gregorianische Kalender**, der im Jahr 1582 eingeführt wurde, **wiederholt sich alle 400 Jahre**. Das bedeutet, dass der 01.01.2024 und der 01.01.2424 beide auf denselben Wochentag fallen. Diese Wiederholung kann genutzt werden, um Berechnungen mit Daten vor 1900 durchzuführen.


Beispiel: Sie wollen den Wochentag für den 01.01.1800 ermitteln. Erfassen Sie das Datum in einer Zelle und addieren Sie 400 Jahre zum ursprünglichen Jahr, um ein Datum zu erhalten, das Excel als solches erkennen kann (Datum nach dem 31.12.1899).

Mit einem Datum vor dem 1.1.1900 arbeiten

Sie können das Datum als Datumswert in der Zelle oder über die Funktion DATUM() erfassen. Für das Beispiel in der folgenden Abbildung lautet die Formel:

=DATUM(Jahr+400; Monat; Tag)

Dabei steht **Jahr** für die Jahreszahl (1800), **Monat** für die Monatsnummer (1 für Januar) und **Tag** für das Tagesdatum (1 für den 1.).




Datum mit der Funktion DATUM() eingeben

Um den Wochentag für den 01.01.1800 zu ermitteln, können Sie jetzt einfach das Zahlenformat in der Zelle ändern, um sich den Wochentag anzeigen zu lassen.

Markieren Sie die Zelle, in der das Datum steht, und drücken Sie die Tastenkombination **Strg + 1**. Es öffnet sich das Dialogfeld **Zellen formatieren**.

Klicken Sie hier auf das Register **Zahlen** und dann am linken Rand auf die Kategorie **Benutzerdefiniert**.


Erfassen Sie unter Typ die Formatzeichen **TTTT** und schließen Sie das Dialogfeld durch einen Klick auf **OK**.



Wochentag eines Datums in Excel anzeigen lassen

Durch die Änderung des Zahlenformats bekommen Sie nun den Namen des Wochentags vom 01.01.1800 angezeigt. Als Alternative hätten Sie auch die Funktion WOCHENTAG() für diese Aufgabenstellung einsetzen können. Also für das Beispiel:

=WOCHENTAG(DATUM(1800+400; 1; 1))



Ausgabe Wochentag für ein Datum vor dem 1.1.1900

Wollen Sie Altersberechnungen oder sonstige Zeitraumberechnungen mit Daten vor dem 01.01.1900 durchführen, dann können Sie analog vorgehen:

- Addieren Sie 400 Jahre zu den entsprechenden Datumswerten.
- Führen Sie Ihre Berechnungen durch.
- Ziehen Sie nach den Berechnungen 400 Jahre ab, um das ursprüngliche Jahr zu erhalten.

Einschränkungen und Hinweise

Ein wichtiger **Hinweis**: Der Gregorianische Kalender wurde 1582 eingeführt. Länder haben ihn zu unterschiedlichen Zeiten übernommen. Stellen Sie daher sicher, dass die Daten im selben Kalendersystem liegen, um genaue Berechnungen zu gewährleisten.

Falls Sie mit Daten vor dem Jahr 1500 arbeiten, müssen Sie 800 Jahre addieren, da die 400-Jahres-Regel nur bis 1500 gilt. Für noch frühere Daten erhöhen Sie die Addition um weitere 400 Jahre je 400-Jahres-Periode (also zum Beispiel 1200, 1600 Jahre, etc.).

Datum im Textformat analysieren und Datum ermitteln

Wenn Sie ein Datum vor dem 1.1.1900 in einer Zelle eingegeben haben und Excel dies als Text erkennt, können Sie mithilfe einer „Textanalyse“ die einzelnen Elemente des Datums entnehmen und mit dem Gregorianischen-Kalender-Trick ein gültiges Datum generieren.

Dazu wird das Text-Datum in Zelle A2 in seine drei Teile zerlegt: Tag, Monat und Jahr. Sie nutzen folgende Funktionen für:

Ausgabe des Tages in Zelle B2:

=LINKS(A2; 2)

Ausgabe des Monats in Zelle C2:

=RECHTS(LINKS(A2; 5); 2)

Ausgabe des Jahres in Zelle D2:

=RECHTS(A2; 4)

Die Ergebnisse setzen Sie dann wie oben beschrieben in ein geeignetes Datum um. Für das Beispiel in der folgenden Abbildung erfassen Sie in Zelle E2:

=DATUM(D2+400; C2; B2)

E3						
=DATUM(D3+400;C3;B3)						
	A	B	C	D	E	F
1	Eingabe	Tag	Monat	Jahr	Datum	Wochentag
2	01.01.1800	01	01	1800	01.01.2200	Mittwoch
3	12.04.1848	12	04	1848	12.04.2248	Mittwoch

Datum vor 1900 zerlegen und wieder zu gültigem Datum zusammenbauen

Etwas komfortabler geht es in Microsoft Excel 365 mit der Array-Funktion in Zelle B2

=TEXTTEILEN(A2; ".")

Das zeigt die folgende Abbildung. Beachten Sie, dass mit der Array-Funktion die Zellen C2 und D2 automatisch mit dem Monat und dem Jahr befüllt werden.

B2						
=TEXTTEILEN(A2;".")						
	A	B	C	D	E	F
1	Eingabe	Tag	Monat	Jahr	Datum	Wochentag
2	12.04.1848	12	04	1848	12.04.2248	Mittwoch

Text-Datum mit TEXTTEILEN() aufschlüsseln

Wenn Sie die Berechnungsspalten B, C, D und E nicht anzeigen wollen, können Sie diese Spalten ausblenden.

Hinweis: Excel und das falsche Schaltjahr 1900

Excel geht fälschlicherweise davon aus, dass das Jahr 1900 ein Schaltjahr ist. Der Fehler besteht schon von Beginn an und wurde bislang nicht korrigiert, weil die Fehlerbeseitigung erhebliche Nachteile für Excel-Nutzer hätte. Alle Formeln, die Datumsangaben nutzen, müssten korrigiert werden. Ein riesiger Aufwand für Excel-Nutzer.

Der Fehler wirkt sich nur auf ein Datum aus, das vor dem 1.3.1900 liegt. Wenn Sie den Wochentag für ein Datum vor dem 1.3.1900 ermitteln wollen, sollten Sie das oben beschriebene Verfahren wählen. Entsprechend gilt für ein Datum vor dem 1.3.1500: Addieren Sie +800 zur Jahreszahl.

Empfehlungen aus dem Management-Handbuch

Mit Excel-Uhrzeit rechnen

Weil Excel die Uhrzeit als Dezimalzahl behandelt, müssen einige Besonderheiten beachtet werden, wenn Sie Uhrzeiten addieren, subtrahieren oder negative Uhrzeiten darstellen wollen. Nutzen Sie diese Tipps und Tricks, um Uhrzeiten korrekt zu berechnen und darzustellen.

<https://www.business-wissen.de/id/kapitel/311/>

Tabellen bearbeiten in Excel

Wie Sie Tabellen für die Datenanalyse vorbereiten und die gewünschten Bereiche auswählen, ergänzen oder kürzen. Mit einer Anleitung für die besondere Excel-Funktion `BEREICH.VERSCHIEBEN` und Funktionen, um Listenwerte zu erzeugen.

<https://www.business-wissen.de/id/kapitel/317/>

Power Pivot - Time Intelligence

Wie Sie die wichtigen Time-Intelligence-Funktionen in Power Pivot nutzen. Mit Anleitungen und Beispielen für `SAMEPERIODLASTYEAR`, `TOTALYTD`, `DATESYTD`, `PARALLELPERIOD` und weitere nützliche DAX-Funktionen.

<https://www.business-wissen.de/id/kapitel/287/>

Nutzen Sie als
Premium-Mitglied
alle
Handbuch-Kapitel
mit mehr als
3.000 Checklisten und Excel-Vorlagen

Jetzt anmelden

www.business-wissen.de/anmelden/

Impressum

b-WISE GmbH Business Wissen Information Service
Bismarckstraße 21
76133 Karlsruhe
DEUTSCHLAND

service@business-wissen.de
Telefon +49 721 18397-0

Copyright 2026, b-wise GmbH, All Rights Reserved